The CNOT Gate 1: Elementary Properties

In[◦]:=
```
Quit[]
```

In[◦]:=
```
Let[Qubit, S]
```

Elementary Properties

In[◦]:=
```
cnot = QuantumCircuit[CNOT[S[1], S[2]]]
```

Out[◦]=



In[◦]:=
```
Matrix[cnot] // MatrixForm
```

Out[◦]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

In[◦]:=
```
in = Basis[S@{1, 2}];
out = cnot ** in
```

Out[◦]=

$$\{ |0_{S_1} 0_{S_2}\rangle, |0_{S_1} 1_{S_2}\rangle, |1_{S_1} 1_{S_2}\rangle, |1_{S_1} 0_{S_2}\rangle \}$$

In[◦]:=
```
ProductForm[Thread[in → out], S@{1, 2}] // TableForm
```

Out[◦]//TableForm=

$$|0\rangle \otimes |0\rangle \to |0\rangle \otimes |0\rangle$$
$$|0\rangle \otimes |1\rangle \to |0\rangle \otimes |1\rangle$$
$$|1\rangle \otimes |0\rangle \to |1\rangle \otimes |1\rangle$$
$$|1\rangle \otimes |1\rangle \to |1\rangle \otimes |0\rangle$$

In[◦]:=
```
Let[Binary, x]
```

In[◦]:=
```
qc = QuantumCircuit[in = Ket[S@{1, 2} → x@{1, 2}], cnot]
```

Out[◦]=



In[ ]:=
```
out = Elaborate[qc]
```
Out[ ]=

$$\left|x_{1_{S_1}}\ x_1\oplus x_{2_{S_2}}\right\rangle$$

In[ ]:=
```
ProductForm[in → out, S@{1, 2}]
```
Out[ ]=

$$\left|x_1\right\rangle\otimes\left|x_2\right\rangle\to\left|x_1\right\rangle\otimes\left|x_1\oplus x_2\right\rangle$$

In summary,

In[ ]:=
```
qc = QuantumCircuit[in = Ket[S@{1, 2} → x@{1, 2}], cnot,
   Ket[S[1] → x[1], S[2] → Mod[x[1] + x[2], 2]],
   "PortSize" → {0.7, 1.5}]
```
Out[ ]=



CNOT copies *basis* states

In[ ]:=
```
Let[Binary, x]
```

In[ ]:=
```
in = Ket[S@{1, 2} → {x, 0}, S@{1, 2}];
qc = QuantumCircuit[in, cnot]
```
Out[ ]=



In[ ]:=
```
out = Elaborate[qc]
```
Out[ ]=

$$\left|x_{S_1}x_{S_2}\right\rangle$$

In[ ]:=
```
ProductForm[in → out, S@{1, 2}]
```
Out[ ]=

$$\left|x\right\rangle\otimes\left|0\right\rangle\to\left|x\right\rangle\otimes\left|x\right\rangle$$

In summary,

In[ ]:=
```
qc = QuantumCircuit[in = Ket[S@{1, 2} → {x, 0}, S@{1, 2}], ,
   Ket[S[1] → x, S[2] → x]]
```
Out[ ]=



No-Cloning Theorem

Quantum mechanics does not allow to copy an unknown state.

$$\left(|0\rangle c_0 + |1\rangle c_1\right) \otimes |0\rangle \mapsto \left(|0\rangle c_0 + |1\rangle c_1\right) \otimes \left(|0\rangle c_0 + |1\rangle c_1\right): \text{NOT ALLOWED!}$$

The result above was possible because the input state was one of the basis states.

So, what happens to a superposition state?

In[•]:=
```
qc1 = QuantumCircuit[Ket[S@{1, 2}], S[1, 6]]
```
Out[•]=



In[•]:=
```
out1 = Elaborate[qc1];
out1 // KetFactor
```
Out[•]=

$$\frac{\left(|0_{S_1}\rangle + |1_{S_1}\rangle\right) \otimes |0_{S_2}\rangle}{\sqrt{2}}$$

In[•]:=
```
qc2 = QuantumCircuit[qc1, "Separator", CNOT[S[1], S[2]]]
```
Out[•]=



In[•]:=
```
out2 = Elaborate[qc2]
```
Out[•]=

$$\frac{|0_{S_1} 0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_1} 1_{S_2}\rangle}{\sqrt{2}}$$

In[•]:=
```
out1 → out2
```
Out[•]=

$$\frac{|0_{S_1} 0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_1} 0_{S_2}\rangle}{\sqrt{2}} \to \frac{|0_{S_1} 0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_1} 1_{S_2}\rangle}{\sqrt{2}}$$

Application: GHZ State

In[ ]:=
```
$n = 3;
CC = S[{0}, $]
TT = S[Range@$n, $]
```
Out[ ]=

$\{S_0\}$

Out[ ]=

$\{S_1, S_2, S_3\}$

In[ ]:=
```
cnot = Map[CNOT[CC, #] &, TT]
```
Out[ ]=

$\{\text{CNOT}[\{S_0\} \to \{1\}, \{S_1\}], \text{CNOT}[\{S_0\} \to \{1\}, \{S_2\}], \text{CNOT}[\{S_0\} \to \{1\}, \{S_3\}]\}$

In[ ]:=
```
qc0 = QuantumCircuit[Ket[CC], Ket[TT], S[0, 6]]
```
Out[ ]=



In[ ]:=
```
out0 = Elaborate[qc0];
out0 // KetFactor
```
Out[ ]=

$$\frac{\left(\left|0_{S_0}\right\rangle + \left|1_{S_0}\right\rangle\right) \otimes \left|0_{S_1} 0_{S_2} 0_{S_3}\right\rangle}{\sqrt{2}}$$

In[ ]:=
```
qc = QuantumCircuit[qc0, "Separator", Sequence @@ cnot]
```
Out[ ]=



In[ ]:=
```
out = Elaborate[qc]
```
Out[ ]=

$$\frac{\left|0_{S_0} 0_{S_1} 0_{S_2} 0_{S_3}\right\rangle}{\sqrt{2}} + \frac{\left|1_{S_0} 1_{S_1} 1_{S_2} 1_{S_3}\right\rangle}{\sqrt{2}}$$

Summary

Functions

- `CNOT`
- `Hadamard`

Related Links

- Chapters 2 of the Quantum Workbook (2022, 2023).
- Tutorial: “Quantum Computation: Overview”