Single-Qubit Rotations

In[ ]:=
```
Let[Qubit, S]
Let[Real, ϕ]
```

Rotation Around the X Axis

In[ ]:=
```
op = Rotation[ϕ, S[1]]
```

Out[ ]=

$\text{Rotation}[\phi, S^x]$

In[ ]:=
```
in = Ket[{S}]
```

Out[ ]=

$\left|0_S\right\rangle$

In[ ]:=
```
out = op ** in
```

Out[ ]=

$\text{Cos}\left[\frac{\phi}{2}\right] \left|0_S\right\rangle - i \left|1_S\right\rangle \text{Sin}\left[\frac{\phi}{2}\right]$

In[ ]:=
```
bv[ϕ_] = BlochVector[out] // ExpToTrig // FullSimplify
```

Out[ ]=

$\{0, -\text{Sin}[\phi], \text{Cos}[\phi]\}$

In[ ]:=
```
Manipulate[BlochSphere[{Red, Bead@bv@ϕ}, ImageSize → Medium], {ϕ, 0, 2 Pi}]
```
Out[ ]=



Rotation Around the Y Axis

In[ ]:=
```
op = Rotation[ϕ, S[2]]
```
Out[ ]=
```
Rotation[ϕ, S^y]
```

In[ ]:=
```
in = Ket[{S}]
```
Out[ ]=

$|0_S\rangle$

In[∘]:= `out = op ** in`

Out[∘]=

$$\text{Cos}\left[\frac{\phi}{2}\right] \; |0_S\rangle + |1_S\rangle \; \text{Sin}\left[\frac{\phi}{2}\right]$$

In[∘]:= `bv[ϕ_] = BlochVector[out] // ExpToTrig // FullSimplify`

Out[∘]=

{Sin[ϕ], 0, Cos[ϕ]}

In[∘]:= `Manipulate[BlochSphere[{Red, Bead@bv@ϕ}, ImageSize → Medium], {ϕ, 0, 2 Pi}]`

Out[∘]=

ϕ

Q3`BlochSphere[{■, Q3`Bead[bv[3.46832]]}, ImageSize → Medium]

Rotation Around the Z Axis

In[∘]:= `op = Rotation[ϕ, S[3]]`

Out[∘]=

Rotation[ϕ, S^z]

In[∘]:= `in = S[6] ** Ket[{S}]`

Out[∘]=

$$\frac{|\_\rangle}{\sqrt{2}} + \frac{|1_S\rangle}{\sqrt{2}}$$

In[∘]:= `out = op ** in`

Out[∘]=

$$\frac{|\_\rangle \left(\text{Cos}\left[\frac{\phi}{2}\right] - i\,\text{Sin}\left[\frac{\phi}{2}\right]\right)}{\sqrt{2}} + \frac{|1_S\rangle \left(\text{Cos}\left[\frac{\phi}{2}\right] + i\,\text{Sin}\left[\frac{\phi}{2}\right]\right)}{\sqrt{2}}$$

In[∘]:= `bv[ϕ_] = BlochVector[out] // ExpToTrig // FullSimplify`

Out[∘]=

{Cos[ϕ], Sin[ϕ], 0}

In[∘]:= `Manipulate[BlochSphere[{Red, Bead@bv@ϕ}, ImageSize → Medium], {ϕ, 0, 2 Pi}]`

Out[∘]=

ϕ

Q3`BlochSphere[{■, Q3`Bead[bv[5.42867]]}, ImageSize → Medium]

Operator Algebra

In[ ]:=
```
op = Rotation[ϕ, S[3]]
```
Out[ ]=

Rotation[ϕ, S^z]

In[ ]:=
```
Elaborate[op]
```
Out[ ]=

$$\text{Cos}\left[\frac{\phi}{2}\right] - i\ S^z\ \text{Sin}\left[\frac{\phi}{2}\right]$$

In[ ]:=
```
SS = S[All]
```
Out[ ]=

{S^x, S^y, S^z}

In[ ]:=
```
TT = op ** SS ** Dagger[op]
```
Out[ ]=

{Cos[ϕ] S^x + S^y Sin[ϕ], Cos[ϕ] S^y - S^x Sin[ϕ], S^z}

In[ ]:=
```
mat = RotationMatrix[ϕ, {0, 0, 1}]
```
Out[ ]=

{{Cos[ϕ], -Sin[ϕ], 0}, {Sin[ϕ], Cos[ϕ], 0}, {0, 0, 1}}

In[ ]:=
```
SS.mat - TT
```
Out[ ]=

{0, 0, 0}

In[ ]:=
```
op = Rotation[ϕ, S[1]]
```
Out[ ]=

Rotation[ϕ, S^x]

In[ ]:=
```
Elaborate[op]
```
Out[ ]=

$$\text{Cos}\left[\frac{\phi}{2}\right] - i\ S^x\ \text{Sin}\left[\frac{\phi}{2}\right]$$

In[ ]:=
```
SS = S[All]
```
Out[ ]=

{S^x, S^y, S^z}

In[ ]:=
```
TT = op ** SS ** Dagger[op]
```
Out[ ]=

{S^x, Cos[ϕ] S^y + S^z Sin[ϕ], Cos[ϕ] S^z - S^y Sin[ϕ]}

In[ ]:=
```
mat = RotationMatrix[ϕ, {1, 0, 0}]
```
Out[ ]=

{{1, 0, 0}, {0, Cos[ϕ], -Sin[ϕ]}, {0, Sin[ϕ], Cos[ϕ]}}

In[ ]:=
```
SS.mat - TT
```
Out[ ]=

{0, 0, 0}

In[ ]:=
```
op = Rotation[ϕ, S[2]]
```
Out[ ]=

$\text{Rotation}[\phi, S^y]$

In[ ]:=
```
Elaborate[op]
```
Out[ ]=

$\text{Cos}\left[\frac{\phi}{2}\right] - i\, S^y\, \text{Sin}\left[\frac{\phi}{2}\right]$

In[ ]:=
```
SS = S[All]
```
Out[ ]=

$\{S^x, S^y, S^z\}$

In[ ]:=
```
TT = op ** SS ** Dagger[op]
```
Out[ ]=

$\{\text{Cos}[\phi]\, S^x - S^z\, \text{Sin}[\phi], S^y, \text{Cos}[\phi]\, S^z + S^x\, \text{Sin}[\phi]\}$

In[ ]:=
```
mat = RotationMatrix[ϕ, {0, 1, 0}]
```
Out[ ]=

$\{\{\text{Cos}[\phi], 0, \text{Sin}[\phi]\}, \{0, 1, 0\}, \{-\text{Sin}[\phi], 0, \text{Cos}[\phi]\}\}$

In[ ]:=
```
SS.mat - TT
```
Out[ ]=

$\{0, 0, 0\}$

Application: Phase and Hadamard

In[ ]:=
```
op = Rotation[ϕ, S[3]]
```
Out[ ]=

$\text{Rotation}[\phi, S^z]$

In[ ]:=
```
mat = Matrix[op];
MatrixForm[mat]
```
Out[ ]//MatrixForm=

$$\begin{pmatrix} e^{-\frac{i\phi}{2}} & 0 \\ 0 & e^{\frac{i\phi}{2}} \end{pmatrix}$$

In[ ]:=
```
Phase[ϕ, S[3]] // Matrix // MatrixForm
```
Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & e^{i\phi} \end{pmatrix}$$

In[ ]:=
```
Exp[I * ϕ / 2] * mat // MatrixForm
```
Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & e^{i\phi} \end{pmatrix}$$

In[ ]:=
```
qc = QuantumCircuit[S[6], Phase[ϕ, S[3]], S[6]]
```
Out[ ]=



In[ ]:=
```
new = QuantumCircuit[Phase[ϕ, S[1]]]
```
Out[ ]=



In[ ]:=
```
qc - new // Elaborate // Simplify
```
Out[ ]=
```
0
```

In[ ]:=
```
qc = QuantumCircuit[S[-7], S[6], Phase[ϕ, S[3]], S[6], S[7]]
```
Out[ ]=



In[ ]:=
```
more = QuantumCircuit[S[-7], Phase[ϕ, S[1]], S[7]]
```
Out[ ]=



In[ ]:=
```
new = QuantumCircuit[Phase[ϕ, S[2]]]
```
Out[ ]=



In[ ]:=
```
qc - new // Elaborate // Simplify
more - new // Elaborate // Simplify
```
Out[ ]=
```
0
```
Out[ ]=
```
0
```

Summary

Functions

- `Rotation`
- `BlochVector, BlochSphere, Bead`
- `Phase`

Related Links

- Chapters 1 and 2 of the Quantum Workbook (2022, 2023).
- Tutorial: “Quantum States”
- Tutorial: “Quantum Computation: Overview”