Phase Operators

```
In[ ]:= Quit[]
```

Half, Quadrant, Octant, Hexadecant

```
In[ ]:= Let[Qubit, S]
```

Let us consider these operators.

```
In[ ]:= ops = S[1, {0, 3, 7, 8, 9}]
```

Out[ ]=

$\{S_1^0, S_1^z, S_1^S, S_1^T, S_1^F\}$

They are diagonal in the computational basis.

```
In[ ]:= MatrixForm /@ Matrix /@ ops
```

Out[ ]=

$\left\{\begin{pmatrix}1 & 0\\ 0 & 1\end{pmatrix}, \begin{pmatrix}1 & 0\\ 0 & -1\end{pmatrix}, \begin{pmatrix}1 & 0\\ 0 & i\end{pmatrix}, \begin{pmatrix}1 & 0\\ 0 & e^{\frac{i\pi}{4}}\end{pmatrix}, \begin{pmatrix}1 & 0\\ 0 & e^{\frac{i\pi}{8}}\end{pmatrix}\right\}$

It means that they do not flip the bit values.

```
In[ ]:= in = Ket[S@{1}]
```

Out[ ]=

$|0_{S_1}\rangle$

```
In[ ]:= out = ops ** in
```

Out[ ]=

$\{|0_{S_1}\rangle, |0_{S_1}\rangle, |0_{S_1}\rangle, |0_{S_1}\rangle, |0_{S_1}\rangle\}$

```
In[ ]:= in = Ket[S[1] → 1]
```

Out[ ]=

$|1_{S_1}\rangle$

```
In[ ]:= out = ops ** in
```

Out[ ]=

$\{|1_{S_1}\rangle, -|1_{S_1}\rangle, i\,|1_{S_1}\rangle, (-1)^{1/4}\,|1_{S_1}\rangle, (-1)^{1/8}\,|1_{S_1}\rangle\}$

Consider the phase operator in the computational basis.

```
In[ ]:= Let[Real, ϕ]
        op = Phase[ϕ, S[1, 3]]
```

Out[ ]=

$S_1^z(\phi)$

In[ ]:=
```
mat = Matrix[op];
MatrixForm[mat]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & e^{i\phi} \end{pmatrix}$$

Then, consider these special angles.

In[ ]:=
```
angles = 2 Pi / HoldForm /@ {1, 2, 4, 8, 16}
```

Out[ ]=

$$\left\{\frac{2\pi}{1}, \frac{2\pi}{2}, \frac{2\pi}{4}, \frac{2\pi}{8}, \frac{2\pi}{16}\right\}$$

In[ ]:=
```
new = Phase[#, S[1, 3]] & /@ angles
MatrixForm /@ Matrix /@ ReleaseHold[new]
```

Out[ ]=

$$\left\{S_1^z\left(\frac{2\pi}{1}\right), S_1^z\left(\frac{2\pi}{2}\right), S_1^z\left(\frac{2\pi}{4}\right), S_1^z\left(\frac{2\pi}{8}\right), S_1^z\left(\frac{2\pi}{16}\right)\right\}$$

Out[ ]=

$$\left\{\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & i \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & \frac{1+i}{\sqrt{2}} \end{pmatrix}, \begin{pmatrix} 1 & 0 \\ 0 & e^{\frac{i\pi}{8}} \end{pmatrix}\right\}$$

Together with the Hadamard

In[ ]:=
```
Let[Qubit, S]
```

In[ ]:=
```
qc = QuantumCircuit[Ket[S@{1}], S[1, 6], Phase[ϕ, S[1, 3]]]
```

Out[ ]=



In[ ]:=
```
in = S[1, 6] ** Ket[]
```

Out[ ]=

$$\frac{|0_{S_1}\rangle}{\sqrt{2}} + \frac{|1_{S_1}\rangle}{\sqrt{2}}$$

In[ ]:=
```
out = Phase[ϕ, S[1, 3]] ** in
```

Out[ ]=

$$\frac{|0_{S_1}\rangle}{\sqrt{2}} + \frac{e^{i\phi}\,|1_{S_1}\rangle}{\sqrt{2}}$$

In[ ]:=
```
out - Elaborate[qc]
```

Out[ ]=

0

Question: What if you want to change the relative amplitude as well?

Phase Shift in the Pauli X Basis

In[ ]:=
```
op = Phase[ϕ, S[1, 1]]
```
Out[ ]=

$$S_1^x(\phi)$$

In[ ]:=
```
in = Ket[S@{1}]
```
Out[ ]=

$$\left|0_{S_1}\right\rangle$$

In[ ]:=
```
out = op ** in
```
Out[ ]=

$$\frac{1}{2}\left(1+e^{i\phi}\right)\left|0_{S_1}\right\rangle+\frac{1}{2}\left(1-e^{i\phi}\right)\left|1_{S_1}\right\rangle$$

Check the input and output states in the X basis.

In[ ]:=
```
XBasisForm[in, S[1]]
```
Out[ ]=

$$\frac{\left|-_{S_1}\right\rangle}{\sqrt{2}}+\frac{\left|+_{S_1}\right\rangle}{\sqrt{2}}$$

In[ ]:=
```
XBasisForm[out, S[1]]
```
Out[ ]=

$$\frac{e^{i\phi}\left|-_{S_1}\right\rangle}{\sqrt{2}}+\frac{\left|+_{S_1}\right\rangle}{\sqrt{2}}$$

Let us further apply the Hadamard.

In[ ]:=
```
new = S[1, 6] ** out
```
Out[ ]=

$$\frac{\left|0_{S_1}\right\rangle}{\sqrt{2}}+\frac{e^{i\phi}\left|1_{S_1}\right\rangle}{\sqrt{2}}$$

In[ ]:=
```
qcZ = QuantumCircuit[Ket[S@{1}], S[1, 6], Phase[ϕ, S[1, 3]]]
```
Out[ ]=



In[ ]:=
```
qcX = QuantumCircuit[Ket[S@{1}], Phase[ϕ, S[1, 1]], S[1, 6]]
```
Out[ ]=



Therefore, we have the identity.

In[ ]:=
```
S[1, 6] ** Phase[ϕ, S[1, 3]] == Phase[ϕ, S[1, 1]] ** S[1, 6]
```
Out[ ]=

$$S_1^H S_1^z(\phi) == S_1^x(\phi)\, S_1^H$$

In[ ]:=
```
S[1, 6] ** Phase[ϕ, S[1, 3]] ** S[1, 6] == Phase[ϕ, S[1, 1]]
```
Out[ ]=
$$S_1^H S_1^z(\phi)\, S_1^H == S_1^x(\phi)$$

In[ ]:=
```
% // Elaborate
```
Out[ ]=
True

Phase Shift in the Pauli Y Basis

In[ ]:=
```
op = Phase[ϕ, S[1, 2]]
```
Out[ ]=
$$S_1^y(\phi)$$

In[ ]:=
```
in = Ket[S@{1}]
```
Out[ ]=
$$\left|0_{S_1}\right\rangle$$

In[ ]:=
```
out = op ** in
```
Out[ ]=
$$\frac{1}{2}\left(1+e^{i\phi}\right)\left|0_{S_1}\right\rangle-\frac{1}{2}\, i\left(-1+e^{i\phi}\right)\left|1_{S_1}\right\rangle$$

Check the input and output states in the X basis.

In[ ]:=
```
YBasisForm[in, S[1]]
```
Out[ ]=
$$\frac{\left|L_{S_1}\right\rangle}{\sqrt{2}}+\frac{\left|R_{S_1}\right\rangle}{\sqrt{2}}$$

In[ ]:=
```
YBasisForm[out, S[1]]
```
Out[ ]=
$$\frac{\left|L_{S_1}\right\rangle}{\sqrt{2}}+\frac{e^{i\phi}\left|R_{S_1}\right\rangle}{\sqrt{2}}$$

Let us further apply the Hadamard.

In[ ]:=
```
new = S[1, 6] ** S[1, 7] ** out
```
Out[ ]=
$$\frac{e^{i\phi}\left|0_{S_1}\right\rangle}{\sqrt{2}}+\frac{\left|1_{S_1}\right\rangle}{\sqrt{2}}$$

In[ ]:=
```
qcZ = QuantumCircuit[Ket[S@{1}], S[1, 6], Phase[ϕ, S[1, 3]]]
```
Out[ ]=


In[∘]:=
```
qcX = QuantumCircuit[Ket[S@{1}], Phase[ϕ, S[1, 2]], S[1, -7], S[1, 6]]
```

Out[∘]=



Therefore, we have the identity.

In[∘]:=
```
HoldForm[S[1, 7]] ** S[1, 6] ** Phase[ϕ, S[1, 3]] ==
 Phase[ϕ, S[1, 2]] ** HoldForm[S[1, 7]] ** S[1, 6]
```

Out[∘]=

$S_1^S\, S_1^H\, S_1^z(\phi) == S_1^y(\phi)\, S_1^S\, S_1^H$

In[∘]:=
```
HoldForm[S[1, 7]] ** S[1, 6] ** Phase[ϕ, S[1, 3]] **
  S[1, 6] ** HoldForm[S[1, -7]] == Phase[ϕ, S[1, 2]]
```

Out[∘]=

$S_1^S\, S_1^H\, S_1^z(\phi)\, S_1^H\, S_1^{S\dagger} == S_1^y(\phi)$

In[∘]:=
```
% // ReleaseHold // Elaborate
```

Out[∘]=

```
True
```

Summary

Functions

- **Phase**
- **XBasisForm, YBasisForm**

Related Links

- Chapters 1 and 2 of the Quantum Workbook (2022, 2023).
- Tutorial: "Quantum States"
- Tutorial: "Quantum Computation: Overview"