Schmidt Decomposition

In[ ]:=
```
Quit[]
```

In[ ]:=
```
Let[Qubit, S]
```

Two Qubits

In[ ]:=
```
bs = Basis[S@{1, 2}]
```

Out[ ]=

$$\{|0_{S_1}0_{S_2}\rangle, |0_{S_1}1_{S_2}\rangle, |1_{S_1}0_{S_2}\rangle, |1_{S_1}1_{S_2}\rangle\}$$

In[ ]:=
```
v = Total@bs
```

Out[ ]=

$$|0_{S_1}0_{S_2}\rangle + |0_{S_1}1_{S_2}\rangle + |1_{S_1}0_{S_2}\rangle + |1_{S_1}1_{S_2}\rangle$$

In[ ]:=
```
KetFactor[v]
```

Out[ ]=

$$(|0_{S_1}\rangle + |1_{S_1}\rangle) \otimes (|0_{S_2}\rangle + |1_{S_2}\rangle)$$

In[ ]:=
```
{val, α, β} = SchmidtDecomposition[v, S[1], S[2]]
```

Out[ ]=

$$\left\{\{2, 0\}, \left\{\frac{|0_{S_1}\rangle}{\sqrt{2}} + \frac{|1_{S_1}\rangle}{\sqrt{2}}, -\frac{|0_{S_1}\rangle}{\sqrt{2}} + \frac{|1_{S_1}\rangle}{\sqrt{2}}\right\}, \left\{\frac{|0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_2}\rangle}{\sqrt{2}}, -\frac{|0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_2}\rangle}{\sqrt{2}}\right\}\right\}$$

In[ ]:=
```
val
```

Out[ ]=

$$\{2, 0\}$$

In[ ]:=
```
α
β
```

Out[ ]=

$$\left\{\frac{|0_{S_1}\rangle}{\sqrt{2}} + \frac{|1_{S_1}\rangle}{\sqrt{2}}, -\frac{|0_{S_1}\rangle}{\sqrt{2}} + \frac{|1_{S_1}\rangle}{\sqrt{2}}\right\}$$

Out[ ]=

$$\left\{\frac{|0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_2}\rangle}{\sqrt{2}}, -\frac{|0_{S_2}\rangle}{\sqrt{2}} + \frac{|1_{S_2}\rangle}{\sqrt{2}}\right\}$$

In[ ]:=
```
new = α[[1]] ** β[[1]] * 2 // Elaborate
```

Out[ ]=

$$|0_{S_1}0_{S_2}\rangle + |0_{S_1}1_{S_2}\rangle + |1_{S_1}0_{S_2}\rangle + |1_{S_1}1_{S_2}\rangle$$

In[ ]:=
```
v - new
```

Out[ ]=

0

In[ ]:= w = Total@Rest[bs]

Out[ ]=

$$|0_{S_1}1_{S_2}\rangle + |1_{S_1}0_{S_2}\rangle + |1_{S_1}1_{S_2}\rangle$$

In[ ]:= KetFactor[w]

Out[ ]=

$$|0_{S_1}1_{S_2}\rangle + |1_{S_1}0_{S_2}\rangle + |1_{S_1}1_{S_2}\rangle$$

In[ ]:= {val, α, β} = SchmidtDecomposition[w, S[1], S[2]];

In[ ]:= val

Out[ ]=

$$\left\{\sqrt{\frac{1}{2}\left(3+\sqrt{5}\right)}, \sqrt{\frac{1}{2}\left(3-\sqrt{5}\right)}\right\}$$

In[ ]:= α // N
β // N

Out[ ]=

$$\left\{0.525731\ |0_{S_1}\rangle + 0.850651\ |1_{S_1}\rangle,\ 0.850651\ |0_{S_1}\rangle - 0.525731\ |1_{S_1}\rangle\right\}$$

Out[ ]=

$$\left\{0.525731\ |0_{S_2}\rangle + 0.850651\ |1_{S_2}\rangle,\ -0.850651\ |0_{S_2}\rangle + 0.525731\ |1_{S_2}\rangle\right\}$$

In[ ]:= new = Total@MapThread[#1 * CircleTimes[#2, #3] &, {val, α, β}]

Out[ ]=

$$\sqrt{\frac{1}{2}\left(3-\sqrt{5}\right)}\left(\frac{1}{10}\left(-5-\sqrt{5}\right)|0_{S_1}0_{S_2}\rangle + \frac{|0_{S_1}1_{S_2}\rangle}{\sqrt{5}} + \frac{|1_{S_1}0_{S_2}\rangle}{\sqrt{5}} + \frac{1}{10}\left(-5+\sqrt{5}\right)|1_{S_1}1_{S_2}\rangle\right) +$$
$$\sqrt{\frac{1}{2}\left(3+\sqrt{5}\right)}\left(\frac{1}{10}\left(5-\sqrt{5}\right)|0_{S_1}0_{S_2}\rangle + \frac{|0_{S_1}1_{S_2}\rangle}{\sqrt{5}} + \frac{|1_{S_1}0_{S_2}\rangle}{\sqrt{5}} + \frac{1}{10}\left(5+\sqrt{5}\right)|1_{S_1}1_{S_2}\rangle\right)$$

In[ ]:= new - w // FullSimplify

Out[ ]=

0

In[∘]:=
```
more = SchmidtForm[w, S[1], S[2]]
```
Out[∘]=

$$\sqrt{\frac{1}{2}\left(3-\sqrt{5}\right)}\left(\frac{\left|0_{S_1}\right\rangle}{\sqrt{1+\left(1+\frac{1}{2}\left(-1-\sqrt{5}\right)\right)^2}}+\frac{\left(1+\frac{1}{2}\left(-1-\sqrt{5}\right)\right)\left|1_{S_1}\right\rangle}{\sqrt{1+\left(1+\frac{1}{2}\left(-1-\sqrt{5}\right)\right)^2}}\right)\otimes\left(\frac{\left(-1-\sqrt{5}\right)\left|0_{S_2}\right\rangle}{2\sqrt{1+\frac{1}{4}\left(-1-\sqrt{5}\right)^2}}+\frac{\left|1_{S_2}\right\rangle}{\sqrt{1+\frac{1}{4}\left(-1-\sqrt{5}\right)^2}}\right)+$$
$$\sqrt{\frac{1}{2}\left(3+\sqrt{5}\right)}\left(\frac{\left|0_{S_1}\right\rangle}{\sqrt{1+\left(1+\frac{1}{2}\left(-1+\sqrt{5}\right)\right)^2}}+\frac{\left(1+\frac{1}{2}\left(-1+\sqrt{5}\right)\right)\left|1_{S_1}\right\rangle}{\sqrt{1+\left(1+\frac{1}{2}\left(-1+\sqrt{5}\right)\right)^2}}\right)\otimes\left(\frac{\left(-1+\sqrt{5}\right)\left|0_{S_2}\right\rangle}{2\sqrt{1+\frac{1}{4}\left(-1+\sqrt{5}\right)^2}}+\frac{\left|1_{S_2}\right\rangle}{\sqrt{1+\frac{1}{4}\left(-1+\sqrt{5}\right)^2}}\right)$$

In[∘]:=
```
more // N // KetRegulate
```
Out[∘]=

$$0.618034\left(0.850651\left|0_{S_1}\right\rangle-0.525731\left|1_{S_1}\right\rangle\right)\otimes\left(-0.850651\left|0_{S_2}\right\rangle+0.525731\left|1_{S_2}\right\rangle\right)+1.61803\left(0.525731\left|0_{S_1}\right\rangle+0.850651\left|1_{S_1}\right\rangle\right)\otimes\left(0.525731\left|0_{S_2}\right\rangle+0.850651\left|1_{S_2}\right\rangle\right)$$

In[∘]:=
```
ReleaseTimes[more]
```
Out[∘]=

$$\sqrt{\frac{1}{2}\left(3-\sqrt{5}\right)}\left(\frac{1}{10}\left(-5-\sqrt{5}\right)\left|0_{S_1}0_{S_2}\right\rangle+\frac{\left|0_{S_1}1_{S_2}\right\rangle}{\sqrt{5}}+\frac{\left|1_{S_1}0_{S_2}\right\rangle}{\sqrt{5}}+\frac{1}{10}\left(-5+\sqrt{5}\right)\left|1_{S_1}1_{S_2}\right\rangle\right)+$$
$$\sqrt{\frac{1}{2}\left(3+\sqrt{5}\right)}\left(\frac{1}{10}\left(5-\sqrt{5}\right)\left|0_{S_1}0_{S_2}\right\rangle+\frac{\left|0_{S_1}1_{S_2}\right\rangle}{\sqrt{5}}+\frac{\left|1_{S_1}0_{S_2}\right\rangle}{\sqrt{5}}+\frac{1}{10}\left(5+\sqrt{5}\right)\left|1_{S_1}1_{S_2}\right\rangle\right)$$

In[∘]:=
```
% - w // FullSimplify
```
Out[∘]=

0

(1+2) Qubits

In[∘]:=
```
bs = Basis[S@{1, 2, 3}]
```
Out[∘]=

$$\left\{\left|0_{S_1}0_{S_2}0_{S_3}\right\rangle,\left|0_{S_1}0_{S_2}1_{S_3}\right\rangle,\left|0_{S_1}1_{S_2}0_{S_3}\right\rangle,\left|0_{S_1}1_{S_2}1_{S_3}\right\rangle,\left|1_{S_1}0_{S_2}0_{S_3}\right\rangle,\left|1_{S_1}0_{S_2}1_{S_3}\right\rangle,\left|1_{S_1}1_{S_2}0_{S_3}\right\rangle,\left|1_{S_1}1_{S_2}1_{S_3}\right\rangle\right\}$$

In[∘]:=
```
v = Total[bs]
```
Out[∘]=

$$\left|0_{S_1}0_{S_2}0_{S_3}\right\rangle+\left|0_{S_1}0_{S_2}1_{S_3}\right\rangle+\left|0_{S_1}1_{S_2}0_{S_3}\right\rangle+\left|0_{S_1}1_{S_2}1_{S_3}\right\rangle+\left|1_{S_1}0_{S_2}0_{S_3}\right\rangle+\left|1_{S_1}0_{S_2}1_{S_3}\right\rangle+\left|1_{S_1}1_{S_2}0_{S_3}\right\rangle+\left|1_{S_1}1_{S_2}1_{S_3}\right\rangle$$

In[∘]:= `KetFactor[v]`

Out[∘]=

$$\left(\left|0_{S_1}\right\rangle + \left|1_{S_1}\right\rangle\right) \otimes \left(\left|0_{S_2}\right\rangle + \left|1_{S_2}\right\rangle\right) \otimes \left(\left|0_{S_3}\right\rangle + \left|1_{S_3}\right\rangle\right)$$

In[∘]:= `{val, α, β} = SchmidtDecomposition[v, S[1], S@{2, 3}]`

Out[∘]=

$$\left\{\left\{2\sqrt{2}, 0\right\}, \left\{\frac{\left|0_{S_1}\right\rangle}{\sqrt{2}} + \frac{\left|1_{S_1}\right\rangle}{\sqrt{2}}, -\frac{\left|0_{S_1}\right\rangle}{\sqrt{2}} + \frac{\left|1_{S_1}\right\rangle}{\sqrt{2}}\right\},\right.$$
$$\left.\left\{\frac{1}{2}\left|0_{S_2}0_{S_3}\right\rangle + \frac{1}{2}\left|0_{S_2}1_{S_3}\right\rangle + \frac{1}{2}\left|1_{S_2}0_{S_3}\right\rangle + \frac{1}{2}\left|1_{S_2}1_{S_3}\right\rangle, -\frac{\left|0_{S_2}0_{S_3}\right\rangle}{\sqrt{2}} + \frac{\left|1_{S_2}1_{S_3}\right\rangle}{\sqrt{2}}\right\}\right\}$$

In[∘]:= `w = Total@Most@bs`

Out[∘]=

$$\left|0_{S_1}0_{S_2}0_{S_3}\right\rangle + \left|0_{S_1}0_{S_2}1_{S_3}\right\rangle + \left|0_{S_1}1_{S_2}0_{S_3}\right\rangle + \left|0_{S_1}1_{S_2}1_{S_3}\right\rangle + \left|1_{S_1}0_{S_2}0_{S_3}\right\rangle + \left|1_{S_1}0_{S_2}1_{S_3}\right\rangle + \left|1_{S_1}1_{S_2}0_{S_3}\right\rangle$$

In[∘]:= `KetFactor[w]`

Out[∘]=

$$\left|0_{S_1}0_{S_2}0_{S_3}\right\rangle + \left|0_{S_1}0_{S_2}1_{S_3}\right\rangle + \left|0_{S_1}1_{S_2}0_{S_3}\right\rangle + \left|0_{S_1}1_{S_2}1_{S_3}\right\rangle + \left|1_{S_1}0_{S_2}0_{S_3}\right\rangle + \left|1_{S_1}0_{S_2}1_{S_3}\right\rangle + \left|1_{S_1}1_{S_2}0_{S_3}\right\rangle$$

In[∘]:= `{val, α, β} = SchmidtDecomposition[w, S[1], S@{2, 3}]`

Out[∘]=

$$\left\{\left\{\sqrt{\frac{1}{2}\left(7+\sqrt{37}\right)}, \sqrt{\frac{1}{2}\left(7-\sqrt{37}\right)}\right\}, \left\{\frac{\left(1+\sqrt{37}\right)\left|0_{S_1}\right\rangle}{6\sqrt{1+\frac{1}{36}\left(1+\sqrt{37}\right)^2}} + \frac{\left|1_{S_1}\right\rangle}{\sqrt{1+\frac{1}{36}\left(1+\sqrt{37}\right)^2}},\right.\right.$$
$$\left.\frac{\left(1-\sqrt{37}\right)\left|0_{S_1}\right\rangle}{6\sqrt{1+\frac{1}{36}\left(1-\sqrt{37}\right)^2}} + \frac{\left|1_{S_1}\right\rangle}{\sqrt{1+\frac{1}{36}\left(1-\sqrt{37}\right)^2}}\right\},$$
$$\left\{\frac{\left(1+\frac{1}{6}\left(1+\sqrt{37}\right)\right)\left|0_{S_2}0_{S_3}\right\rangle}{\sqrt{\frac{1}{36}\left(1+\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1+\sqrt{37}\right)\right)^2}} + \frac{\left(1+\frac{1}{6}\left(1+\sqrt{37}\right)\right)\left|0_{S_2}1_{S_3}\right\rangle}{\sqrt{\frac{1}{36}\left(1+\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1+\sqrt{37}\right)\right)^2}} +\right.$$
$$\frac{\left(1+\frac{1}{6}\left(1+\sqrt{37}\right)\right)\left|1_{S_2}0_{S_3}\right\rangle}{\sqrt{\frac{1}{36}\left(1+\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1+\sqrt{37}\right)\right)^2}} + \frac{\left(1+\sqrt{37}\right)\left|1_{S_2}1_{S_3}\right\rangle}{6\sqrt{\frac{1}{36}\left(1+\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1+\sqrt{37}\right)\right)^2}},$$
$$\frac{\left(1+\frac{1}{6}\left(1-\sqrt{37}\right)\right)\left|0_{S_2}0_{S_3}\right\rangle}{\sqrt{\frac{1}{36}\left(1-\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1-\sqrt{37}\right)\right)^2}} + \frac{\left(1+\frac{1}{6}\left(1-\sqrt{37}\right)\right)\left|0_{S_2}1_{S_3}\right\rangle}{\sqrt{\frac{1}{36}\left(1-\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1-\sqrt{37}\right)\right)^2}} +$$
$$\left.\left.\frac{\left(1+\frac{1}{6}\left(1-\sqrt{37}\right)\right)\left|1_{S_2}0_{S_3}\right\rangle}{\sqrt{\frac{1}{36}\left(1-\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1-\sqrt{37}\right)\right)^2}} + \frac{\left(1-\sqrt{37}\right)\left|1_{S_2}1_{S_3}\right\rangle}{6\sqrt{\frac{1}{36}\left(1-\sqrt{37}\right)^2+3\left(1+\frac{1}{6}\left(1-\sqrt{37}\right)\right)^2}}\right\}\right\}$$

In[∘]:= `val`

Out[∘]=

$$\left\{\sqrt{\frac{1}{2}\left(7+\sqrt{37}\right)}, \sqrt{\frac{1}{2}\left(7-\sqrt{37}\right)}\right\}$$

In[∘]:=
```
more = SchmidtForm[w, S[1], S@{2, 3}] // N
```
Out[∘]=

$$2.55761\ \left(0.76302\ \left|0_{S_1}\right\rangle + 0.646375\ \left|1_{S_1}\right\rangle\right) \otimes \left(0.551059\ \left|0_{S_2}0_{S_3}\right\rangle + 0.551059\ \left|0_{S_2}1_{S_3}\right\rangle + 0.551059\ \left|1_{S_2}0_{S_3}\right\rangle + 0.298333\ \left|1_{S_2}1_{S_3}\right\rangle\right) + 0.677214\ \left(-0.646375\ \left|0_{S_1}\right\rangle + 0.76302\ \left|1_{S_1}\right\rangle\right) \otimes \left(0.172243\ \left|0_{S_2}0_{S_3}\right\rangle + 0.172243\ \left|0_{S_2}1_{S_3}\right\rangle + 0.172243\ \left|1_{S_2}0_{S_3}\right\rangle - 0.954462\ \left|1_{S_2}1_{S_3}\right\rangle\right)$$

In[∘]:=
```
new = ReleaseTimes[more]
```
Out[∘]=

$$0.677214\ \left(-0.111333\ \left|0_{S_1}0_{S_2}0_{S_3}\right\rangle - 0.111333\ \left|0_{S_1}0_{S_2}1_{S_3}\right\rangle - 0.111333\ \left|0_{S_1}1_{S_2}0_{S_3}\right\rangle + 0.61694\ \left|0_{S_1}1_{S_2}1_{S_3}\right\rangle + 0.131425\ \left|1_{S_1}0_{S_2}0_{S_3}\right\rangle + 0.131425\ \left|1_{S_1}0_{S_2}1_{S_3}\right\rangle + 0.131425\ \left|1_{S_1}1_{S_2}0_{S_3}\right\rangle - 0.728273\ \left|1_{S_1}1_{S_2}1_{S_3}\right\rangle\right) + 2.55761\ \left(0.420469\ \left|0_{S_1}0_{S_2}0_{S_3}\right\rangle + 0.420469\ \left|0_{S_1}0_{S_2}1_{S_3}\right\rangle + 0.420469\ \left|0_{S_1}1_{S_2}0_{S_3}\right\rangle + 0.227634\ \left|0_{S_1}1_{S_2}1_{S_3}\right\rangle + 0.356191\ \left|1_{S_1}0_{S_2}0_{S_3}\right\rangle + 0.356191\ \left|1_{S_1}0_{S_2}1_{S_3}\right\rangle + 0.356191\ \left|1_{S_1}1_{S_2}0_{S_3}\right\rangle + 0.192835\ \left|1_{S_1}1_{S_2}1_{S_3}\right\rangle\right)$$

In[∘]:=
```
new - w // Garner // Chop
```
Out[∘]=

0

Entanglement, and so what?

In[∘]:=
```
Let[Qubit, S]
```

In[∘]:=
```
v = Ket[] + Ket[S@{1, 2} → 1] // KetRegulate
```
Out[∘]=

$$\left|0_{S_1}0_{S_2}\right\rangle + \left|1_{S_1}1_{S_2}\right\rangle$$

In[∘]:=
```
qc1 = QuantumCircuit[v,
  Measurement[S[2, 3]],
  "PortSize" → {1.65, 0.5}]
```
Out[∘]=



In[∘]:=
```
data = Table[Elaborate[qc1]; Readout[S[2, 3]], 20]
```
Out[∘]=

{1, 0, 1, 1, 0, 1, 1, 1, 0, 0, 1, 0, 1, 1, 1, 0, 0, 1, 0, 1}

In[ ]:=
```
EchoTiming[data = Table[Elaborate[qc1]; Readout[S[2, 3]], 1000];]
Histogram[data, ImageSize → Medium]
```

2.33801

Out[ ]=



In[ ]:=
```
qc2 = QuantumCircuit[v,
  Measurement[S[1, 3]], "Separator",
  Measurement[S[2, 3]],
  "Invisible" → S@{1.5},
  "PortSize" → {1.65, 0.5}]
```

Out[ ]=



In[ ]:=
```
data = Table[Elaborate[qc2]; Readout[S[{1, 2}, 3]], 20]
```

Out[ ]=
```
{{0, 0}, {1, 1}, {0, 0}, {0, 0}, {0, 0}, {0, 0}, {0, 0}, {1, 1}, {1, 1}, {0, 0},
 {0, 0}, {0, 0}, {0, 0}, {1, 1}, {0, 0}, {1, 1}, {0, 0}, {0, 0}, {1, 1}, {1, 1}}
```

```
In[ ]:= EchoTiming[data = Table[Elaborate[qc2];
          Readout[S[{1, 2}, 3]], 1000];]
        Histogram3D[data, ImageSize → Medium]

        3.56709
Out[ ]=
```



Summary

Functions

- **SchmidtDecomposition**
- **SchmidtForm**

Related Links

- Chapters 1 and 2 of the Quantum Workbook (2022, 2023).
- Tutorial: “Quantum States”
- Tutorial: “Quantum Computation: Overview”