The CNOT Gate 2: Parity Measurement

In[◦]:=
```
Let[Qubit, S]
```

What is Parity?

Parity := $Z_1 Z_2 \ldots Z_n$, where $Z_k \in \{1, -1\}$. Why?

Let us start with a single qubit.

In[◦]:=
```
S[1, 3] // PauliForm
```
Out[◦]=

Z

In[◦]:=
```
in = Basis[S[1]]
```
Out[◦]=

$\{|0_{S_1}\rangle, |1_{S_1}\rangle\}$

In[◦]:=
```
out = S[1, 3] ** in
```
Out[◦]=

$\{|0_{S_1}\rangle, -|1_{S_1}\rangle\}$

In[◦]:=
```
Thread[in → out] // TableForm
```
Out[◦]//TableForm=

$|0_{S_1}\rangle \to |0_{S_1}\rangle$

$|1_{S_1}\rangle \to -|1_{S_1}\rangle$

Now, consider the two-qubit case.

In[◦]:=
```
S[1, 3] ** S[2, 3] // PauliForm
```
Out[◦]=

$Z \otimes Z$

In[◦]:=
```
in = Basis[S@{1, 2}];
out = S[1, 3] ** S[2, 3] ** in;
Thread[in → out] // TableForm
```
Out[◦]//TableForm=

$|0_{S_1}0_{S_2}\rangle \to |0_{S_1}0_{S_2}\rangle$

$|0_{S_1}1_{S_2}\rangle \to -|0_{S_1}1_{S_2}\rangle$

$|1_{S_1}0_{S_2}\rangle \to -|1_{S_1}0_{S_2}\rangle$

$|1_{S_1}1_{S_2}\rangle \to |1_{S_1}1_{S_2}\rangle$

```
In[ ]:= S[1, 3] ** S[2, 3] ** S[3, 3] // PauliForm
Out[ ]=
```

$Z \otimes Z \otimes Z$

```
In[ ]:= in = Basis[S@{1, 2, 3}];
        out = S[1, 3] ** S[2, 3] ** S[3, 3] ** in;
        Thread[in → out] // TableForm
Out[ ]//TableForm=
```

$$\begin{array}{l}
\left|0_{S_1}0_{S_2}0_{S_3}\right\rangle \to \left|0_{S_1}0_{S_2}0_{S_3}\right\rangle \\
\left|0_{S_1}0_{S_2}1_{S_3}\right\rangle \to -\left|0_{S_1}0_{S_2}1_{S_3}\right\rangle \\
\left|0_{S_1}1_{S_2}0_{S_3}\right\rangle \to -\left|0_{S_1}1_{S_2}0_{S_3}\right\rangle \\
\left|0_{S_1}1_{S_2}1_{S_3}\right\rangle \to \left|0_{S_1}1_{S_2}1_{S_3}\right\rangle \\
\left|1_{S_1}0_{S_2}0_{S_3}\right\rangle \to -\left|1_{S_1}0_{S_2}0_{S_3}\right\rangle \\
\left|1_{S_1}0_{S_2}1_{S_3}\right\rangle \to \left|1_{S_1}0_{S_2}1_{S_3}\right\rangle \\
\left|1_{S_1}1_{S_2}0_{S_3}\right\rangle \to \left|1_{S_1}1_{S_2}0_{S_3}\right\rangle \\
\left|1_{S_1}1_{S_2}1_{S_3}\right\rangle \to -\left|1_{S_1}1_{S_2}1_{S_3}\right\rangle
\end{array}$$

```
In[ ]:= Let[Binary, x]
```

```
In[ ]:= kk = Range[4];
        SS = S[kk, $];
        xx = x@kk;
        ProductForm[Ket[SS → xx] → Power[-1, Total@xx] Ket[SS → xx], SS]
Out[ ]=
```

$$|x_1\rangle \otimes |x_2\rangle \otimes |x_3\rangle \otimes |x_4\rangle \to (-1)^{x_1+x_2+x_3+x_4}\ |x_1\rangle \otimes |x_2\rangle \otimes |x_3\rangle \otimes |x_4\rangle$$

```
In[ ]:= ProductForm[Ket[SS → xx] → Power[-1, Mod[Total@xx, 2]] Ket[SS → xx], SS]
Out[ ]=
```

$$|x_1\rangle \otimes |x_2\rangle \otimes |x_3\rangle \otimes |x_4\rangle \to (-1)^{x_1\oplus x_2\oplus x_3\oplus x_4}\ |x_1\rangle \otimes |x_2\rangle \otimes |x_3\rangle \otimes |x_4\rangle$$

Summary

- $Z_1 Z_2 \ldots Z_n \leftrightarrow x_1 \oplus x_2 \oplus \ldots \oplus x_n$, where $x_k \in \{0, 1\}$.
- Binary representation: $\begin{cases} +1 \leftrightarrow 0 \\ -1 \leftrightarrow 1 \end{cases}$

Parity Measurement in Q3

Q3 directly supports the parity measurement.
Consider two quantum registers; native and ancillary registers.

```
In[ ]:= Let[Qubit, S]
```

In[∘]:=
```
$n = 4;
kk = Range[$n];
SS = S[Range@$n, None]
```

Out[∘]=

$\{S_1, S_2, S_3, S_4\}$

In[∘]:=
```
$N = Power[2, $n];
ff = Table[Exp[2 * Pi / $N * HoldForm@Evaluate[k * I]], {k, 0, $N - 1}]
```

Out[∘]=

$$\left\{e^{\frac{\pi 0}{8}}, e^{\frac{\pi i}{8}}, e^{\frac{1}{8}\pi(2 i)}, e^{\frac{1}{8}\pi(3 i)}, e^{\frac{1}{8}\pi(4 i)}, e^{\frac{1}{8}\pi(5 i)}, e^{\frac{1}{8}\pi(6 i)}, e^{\frac{1}{8}\pi(7 i)}, e^{\frac{1}{8}\pi(8 i)}, e^{\frac{1}{8}\pi(9 i)}, e^{\frac{1}{8}\pi(10 i)}, e^{\frac{1}{8}\pi(11 i)}, e^{\frac{1}{8}\pi(12 i)}, e^{\frac{1}{8}\pi(13 i)}, e^{\frac{1}{8}\pi(14 i)}, e^{\frac{1}{8}\pi(15 i)}\right\}$$

In[∘]:=
```
qc = QuantumCircuit["Spacer", Measurement[Multiply @@ S[kk, 3]]]
```

Out[∘]=



Take a superposition of all computational basis states.

In[∘]:=
```
in = Basis[SS].ff;
in // SimpleForm
```

Out[∘]=

$$e^{\frac{\pi 0}{8}} |0000\rangle + e^{\frac{\pi i}{8}} |0001\rangle + e^{\frac{1}{8}\pi(2 i)} |0010\rangle + e^{\frac{1}{8}\pi(3 i)} |0011\rangle + e^{\frac{1}{8}\pi(4 i)} |0100\rangle + e^{\frac{1}{8}\pi(5 i)} |0101\rangle + e^{\frac{1}{8}\pi(6 i)} |0110\rangle + e^{\frac{1}{8}\pi(7 i)} |0111\rangle + e^{\frac{1}{8}\pi(8 i)} |1000\rangle + e^{\frac{1}{8}\pi(9 i)} |1001\rangle + e^{\frac{1}{8}\pi(10 i)} |1010\rangle + e^{\frac{1}{8}\pi(11 i)} |1011\rangle + e^{\frac{1}{8}\pi(12 i)} |1100\rangle + e^{\frac{1}{8}\pi(13 i)} |1101\rangle + e^{\frac{1}{8}\pi(14 i)} |1110\rangle + e^{\frac{1}{8}\pi(15 i)} |1111\rangle$$

Check the output state.

In[∘]:=
```
Quiet[out = qc ** in, Measurement::nonum];
SimpleForm[out, SS]
```

Out[∘]=

$$\frac{(-1)^{1/8} |0001\rangle}{2\sqrt{2}} + \left(\frac{1}{4} + \frac{i}{4}\right) |0010\rangle + \frac{i |0100\rangle}{2\sqrt{2}} + \frac{(-1)^{7/8} |0111\rangle}{2\sqrt{2}} - \frac{|1000\rangle}{2\sqrt{2}} - \frac{(-1)^{3/8} |1011\rangle}{2\sqrt{2}} - \frac{(-1)^{5/8} |1101\rangle}{2\sqrt{2}} + \left(\frac{1}{4} - \frac{i}{4}\right) |1110\rangle$$

Question: Real quantum computers support only measurement Pauli Z of individual qubits. How can we achieve the parity measurement?

Elementary Properties

```
In[◦]:= Let[Binary, x]
```

```
In[◦]:= cnot = QuantumCircuit[CNOT[S[1], S[2]]]
Out[◦]=
```



```
In[◦]:= qc = QuantumCircuit[in = Ket[S@{1, 2} → x@{1, 2}], cnot,
          Ket[S[1] → x[1], S[2] → Mod[x[1] + x[2], 2]],
          "PortSize" → {0.7, 1.5}]
Out[◦]=
```



Application:Parity Measurement

- We want to measure $Z_1 Z_2 \ldots Z_n$, where $Z_k \in \{1, -1\}$.
- We fist note that $Z_1 Z_2 \ldots Z_n \leftrightarrow x_1 \oplus x_2 \oplus \ldots \oplus x_n$, where $x_k \in \{0, 1\}$.

Consider two quantum registers; native and ancillary registers.

```
In[◦]:= Let[Qubit, S, T]
```

```
In[◦]:= $n = 4;
        kk = Range[$n];
        SS = S[Range@$n, None]
Out[◦]=
        {S1, S2, S3, S4}
```

```
In[◦]:= $N = Power[2, $n];
        ff = Table[Exp[I * HoldForm@Evaluate[2 * Pi * k / $N]], {k, 0, $N - 1}];
```

In[ ]:=
```
qc = QuantumCircuit[Ket[{T}], Sequence @@ Map[CNOT[#, T] &, SS],
  "Spacer", Measurement[T[3]], "Invisible" → S[$n + 1 / 2]]
```
Out[ ]=



Take a superposition of all computational basis states.

In[ ]:=
```
in = Basis[SS].ff;
in // SimpleForm
```
Out[ ]=

$$e^{i\,0}\,|0000\rangle + e^{i\frac{\pi}{8}}\,|0001\rangle + e^{i\frac{\pi}{4}}\,|0010\rangle + e^{i\frac{3\pi}{8}}\,|0011\rangle + e^{i\frac{\pi}{2}}\,|0100\rangle + e^{i\frac{5\pi}{8}}\,|0101\rangle + e^{i\frac{3\pi}{4}}\,|0110\rangle + e^{i\frac{7\pi}{8}}\,|0111\rangle + e^{i\pi}\,|1000\rangle + e^{i\frac{9\pi}{8}}\,|1001\rangle + e^{i\frac{5\pi}{4}}\,|1010\rangle + e^{i\frac{11\pi}{8}}\,|1011\rangle + e^{i\frac{3\pi}{2}}\,|1100\rangle + e^{i\frac{13\pi}{8}}\,|1101\rangle + e^{i\frac{7\pi}{4}}\,|1110\rangle + e^{i\frac{15\pi}{8}}\,|1111\rangle$$

Check the output state.

In[ ]:=
```
Quiet[out = qc ** in, Measurement::nonum];
SimpleForm[out, {SS, T}]
```
Out[ ]=

$$\frac{|0000;0\rangle}{2\sqrt{2}} + \frac{(-1)^{3/8}\,|0011;0\rangle}{2\sqrt{2}} + \frac{(-1)^{5/8}\,|0101;0\rangle}{2\sqrt{2}} - \left(\frac{1}{4} - \frac{i}{4}\right)|0110;0\rangle - \frac{(-1)^{1/8}\,|1001;0\rangle}{2\sqrt{2}} - \left(\frac{1}{4} + \frac{i}{4}\right)|1010;0\rangle - \frac{i\,|1100;0\rangle}{2\sqrt{2}} - \frac{(-1)^{7/8}\,|1111;0\rangle}{2\sqrt{2}}$$

In[ ]:=
```
new = QuantumCircuit["Spacer", Measurement[Multiply @@ S[kk, 3]]]
```
Out[ ]=



In[ ]:=
```
Quiet[out = new ** in, Measurement::nonum];
SimpleForm[out]
```

Out[ ]=

$$\frac{|0000\rangle}{2\sqrt{2}} + \frac{(-1)^{3/8}\ |0011\rangle}{2\sqrt{2}} + \frac{(-1)^{5/8}\ |0101\rangle}{2\sqrt{2}} - \left(\frac{1}{4} - \frac{i}{4}\right)\ |0110\rangle - \frac{(-1)^{1/8}\ |1001\rangle}{2\sqrt{2}} - \left(\frac{1}{4} + \frac{i}{4}\right)\ |1010\rangle - \frac{i\ |1100\rangle}{2\sqrt{2}} - \frac{(-1)^{7/8}\ |1111\rangle}{2\sqrt{2}}$$

Summary

Functions

- **CNOT**
- **Hadamard**

Related Links

- Chapters 2 of the Quantum Workbook (2022, 2023).
- Tutorial: “Quantum Computation: Overview”