Multi-Control NOT Gate

In[∘]:=
```
Let[Qubit, S]
```

CNOT: Review

In[∘]:=
```
cnot = QuantumCircuit[CNOT[S[1], S[2]]]
```

Out[∘]=



In[∘]:=
```
in = Basis[S@{1, 2}];
out = cnot ** in;
ProductForm[Thread[in → out], S@{1, 2}] // TableForm
```

Out[∘]//TableForm=

$|0\rangle \otimes |0\rangle \rightarrow |0\rangle \otimes |0\rangle$
$|0\rangle \otimes |1\rangle \rightarrow |0\rangle \otimes |1\rangle$
$|1\rangle \otimes |0\rangle \rightarrow |1\rangle \otimes |1\rangle$
$|1\rangle \otimes |1\rangle \rightarrow |1\rangle \otimes |0\rangle$

In[∘]:=
```
Let[Binary, x]
```

In[∘]:=
```
qc = QuantumCircuit[in = Ket[S@{1, 2} → x@{1, 2}], cnot,
   Ket[S[1] → x[1], S[2] → Mod[x[1] + x[2], 2]],
   "PortSize" → {0.7, 1.5}]
```

Out[∘]=



CNOT on the opposite condition

In[∘]:=
```
xnot = QuantumCircuit[S[1, 1], CNOT[S[1], S[2]], S[1, 1]]
```

Out[∘]=



In[ ]:=
```
in = Basis[S@{1, 2}];
out = xnot ** in;
ProductForm[Thread[in → out], S@{1, 2}] // TableForm
```

Out[ ]//TableForm=

$|0\rangle \otimes |0\rangle \to |0\rangle \otimes |1\rangle$

$|0\rangle \otimes |1\rangle \to |0\rangle \otimes |0\rangle$

$|1\rangle \otimes |0\rangle \to |1\rangle \otimes |0\rangle$

$|1\rangle \otimes |1\rangle \to |1\rangle \otimes |1\rangle$

In[ ]:=
```
TableForm[{MatrixForm /@ Matrix /@ {cnot, xnot}},
 TableHeadings → {None, {CNOT, XNOT}},
 TableAlignments → Center]
```

Out[ ]//TableForm=

CNOT	XNOT
$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}$	$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$

In[ ]:=
```
CNOT[S[1], S[2]]
```

Out[ ]=

CNOT[$\{S_1\} \to \{1\}$, $\{S_2\}$]

In[ ]:=
```
CNOT[S[1] → 0, S[2]]
```

Out[ ]=

CNOT[$\{S_1\} \to \{0\}$, $\{S_2\}$]

In[ ]:=
```
new = QuantumCircuit[CNOT[S[1] → 0, S[2]]]
```

Out[ ]=



In[ ]:=
```
new - xnot // Elaborate
```

Out[ ]=

0

In[ ]:=
```
Let[Binary, x]
```

In[ ]:=
```
qc = QuantumCircuit[in = Ket[S@{1, 2} → x@{1, 2}], new,
   Ket[S[1] → x[1], S[2] → Mod[1 + x[1] + x[2], 2]],
   "PortSize" → {0.7, 2}]
```

Out[ ]=



Multi-Control-NOT Gate

In[∘]:=
```
QuantumCircuit[CNOT[S[{1, 2, 3}, $], S[4]]]
```
Out[∘]=



In[∘]:=
```
$n = 2;
cc = S[Range[$n], $];
tt = S[$n + 1, $];
cnot = QuantumCircuit[CNOT[cc, tt]]
```
Out[∘]=



In[∘]:=
```
Matrix[cnot] // MatrixForm
```
Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \end{pmatrix}$$

In[∘]:=
```
in = Basis[cc, tt];
out = cnot ** in;
```

In[ ]:=
```
ProductForm[Thread[in → out], {cc, tt}] // TableForm
```
Out[ ]//TableForm=

$|00\rangle \otimes |0\rangle \to |00\rangle \otimes |0\rangle$
$|00\rangle \otimes |1\rangle \to |00\rangle \otimes |1\rangle$
$|01\rangle \otimes |0\rangle \to |01\rangle \otimes |0\rangle$
$|01\rangle \otimes |1\rangle \to |01\rangle \otimes |1\rangle$
$|10\rangle \otimes |0\rangle \to |10\rangle \otimes |0\rangle$
$|10\rangle \otimes |1\rangle \to |10\rangle \otimes |1\rangle$
$|11\rangle \otimes |0\rangle \to |11\rangle \otimes |1\rangle$
$|11\rangle \otimes |1\rangle \to |11\rangle \otimes |0\rangle$

In[ ]:=
```
CNOT[S[{1, 2}, $], S[3]]
```
Out[ ]=

$\mathrm{CNOT}[\{S_1, S_2\} \to \{1, 1\}, \{S_3\}]$

In[ ]:=
```
CNOT[S[{1, 2}, $] → {1, 0}, S[3]]
```
Out[ ]=

$\mathrm{CNOT}[\{S_1, S_2\} \to \{1, 0\}, \{S_3\}]$

In[ ]:=
```
qc = QuantumCircuit[CNOT[S@{1, 2} → {1, 0}, S[3]]]
```
Out[ ]=



In[ ]:=
```
in = Basis[S@{1, 2, 3}];
out = qc ** in;
```

In[ ]:=
```
ProductForm[Thread[in → out], {S@{1, 2}, S[3]}] // TableForm
```
Out[ ]//TableForm=

$|00\rangle \otimes |0\rangle \to |00\rangle \otimes |0\rangle$
$|00\rangle \otimes |1\rangle \to |00\rangle \otimes |1\rangle$
$|01\rangle \otimes |0\rangle \to |01\rangle \otimes |0\rangle$
$|01\rangle \otimes |1\rangle \to |01\rangle \otimes |1\rangle$
$|10\rangle \otimes |0\rangle \to |10\rangle \otimes |1\rangle$
$|10\rangle \otimes |1\rangle \to |10\rangle \otimes |0\rangle$
$|11\rangle \otimes |0\rangle \to |11\rangle \otimes |0\rangle$
$|11\rangle \otimes |1\rangle \to |11\rangle \otimes |1\rangle$

Summary

Functions

- `CNOT`
- `Hadamard`

Related Links

- Chapters 2 of the Quantum Workbook (2022, 2023).
- Tutorial: “Quantum Computation: Overview”