QUANTUM PLAYBOOK TECH NOTE

URL

Entanglement Distillation

NEW IN 13.1

See also Section 7.3 of the Quantum Workbook (Springer, 2022) .

Here, we start with a number of *partially* entangled pairs and "distills" a smaller number of *maximally* entangled pairs. This is illustrated in the quantum circuit model.

VonNeumannEntropy	VonNeumann entropy of a mixed state
PartialTrace	Partial trace over some subsystems
Dyad	Dyadic product of two vectors

Functions used in this document.

Make sure that Q3 is loaded.

```
In[3]:= << Q3`
```

We have two parties Alice (A) and Bob (B). Teddy (T) is an ancillary register to perform a measurement of the total Pauli Z.

```
In[4]:= Let[Qubit, A, B, T]
```

We will start with *n* partially entangled pairs.

```
In[5]:= $n = 4;
kk = Range[$n];
AA = A[kk, None];
BB = B[kk, None];
```

You need fewer qubits for T, just enough to store a number up to maximum *n*.

```
In[9]:= ln = Ceiling@Log[2, $n + 1];
TT = T[Range@ln, None];
```

To Generate Partially Entangled Pairs

A Single Pair

Let us first examine a single pair of partially entangled qubits.

First, construct a superposition of the two computational basis states with different amplitudes.

In[39]:=

```
Let[Real, ϕ]
qc = QuantumCircuit[
  LogicalForm[Ket[], A[1]],
  Rotation[ϕ, A[1, 2]]]
```

Out[40]=



Parameter ϕ tunes the two amplitudes, $c_0 = \cos(\phi/2)$ and $c_1 = \sin(\phi/2)$, in the superposition.

In[41]:=

```
Elaborate[qc] // ExpandAll // LogicalForm
```

Out[41]=

$$\mathrm{Cos}\left[\frac{\phi}{2}\right] \left|0_{A_1}\right\rangle + \left|1_{A_1}\right\rangle \mathrm{Sin}\left[\frac{\phi}{2}\right]$$

Then, by applying the CNOT gate, you can create a partially entangled pair.

In[19]:=

```
qc = QuantumCircuit[
  Ket[{A[1], B[1]}],
  Rotation[ϕ, A[1, 2]],
  CNOT[A[1], B[1]]]
```

Out[19]=



Eventually, we see that the parameter ϕ tunes the extent of entanglement (with $\phi = \pi/2$ corresponds to the maximal entanglement).

In[43]:=

```
out = Elaborate[qc];
LogicalForm[out]
```

Out[44]=

$$\mathrm{Cos}\left[\frac{\phi}{2}\right] \left|0_{A_1} 0_{B_1}\right\rangle + \left|1_{A_1} 1_{B_1}\right\rangle \mathrm{Sin}\left[\frac{\phi}{2}\right]$$

Examine the reduced density matrix for the first qubit to check that the above state is partial entangled.

In[149]:=

```
rho = PartialTrace[out, B[1]] // ExpToTrig // Simplify;
Matrix[rho] // ExpToTrig // SimplifyThrough // MatrixForm
```

Out[150]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}\left[\frac{\phi}{2}\right]^2 & 0 \\ 0 & \mathrm{Sin}\left[\frac{\phi}{2}\right]^2 \end{pmatrix}$$

Multiple Pairs

Now, we consider n partially entangled pairs with each pair in the state described above. Let us take a look at the explicit expression for the n partially entangled pairs. Here, OTimes is used just for better readability, and you can replace it with CircleTimes (⊗) or Multiply to get exactly the same result.

Repeating the above procedure for other pairs, generate as many partially entangled pairs as you like. In this particular example, we prepare four pairs.

In[152]:=

```
qc0 = QuantumCircuit[
  LogicalForm[Ket[], AA],
  LogicalForm[Ket[], BB],
  Rotation[ϕ, A[kk, 2]],
  Sequence @@ ReleaseHold@Thread@Hold[CNOT][AA, BB],
  "Invisible" -> A[$n + 1 / 2]
 ]
```

Out[152]=



Expanding it, one gets various terms, each with identical parts on Alice's and Bob's side.

In[156]:=

```
out = Elaborate[qc0];
ProductForm[out, {AA, BB}]
```

Out[157]=

$$\mathrm{Cos}\left[\frac{\phi}{2}\right]^4 |0000\rangle\otimes|0000\rangle + \mathrm{Cos}\left[\frac{\phi}{2}\right]^3 |0001\rangle\otimes|0001\rangle\,\mathrm{Sin}\left[\frac{\phi}{2}\right] + |1111\rangle\otimes|1111\rangle\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^4 +$$
$$\frac{1}{2}\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 |0100\rangle\otimes|0100\rangle\,\mathrm{Sin}[\phi] + \frac{1}{2}\mathrm{Cos}\left[\frac{\phi}{2}\right]^2 |1000\rangle\otimes|1000\rangle\,\mathrm{Sin}[\phi] + \frac{1}{2}|0111\rangle\otimes|0111\rangle\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2\mathrm{Sin}[\phi] +$$
$$\frac{1}{2}|1011\rangle\otimes|1011\rangle\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2\mathrm{Sin}[\phi] + \frac{1}{2}|1101\rangle\otimes|1101\rangle\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2\mathrm{Sin}[\phi] + \frac{1}{2}|1110\rangle\otimes|1110\rangle\,\mathrm{Sin}\left[\frac{\phi}{2}\right]^2\mathrm{Sin}[\phi] +$$
$$\frac{1}{4}\mathrm{Cot}\left[\frac{\phi}{2}\right] |0010\rangle\otimes|0010\rangle\,\mathrm{Sin}[\phi]^2 + \frac{1}{4}|0011\rangle\otimes|0011\rangle\,\mathrm{Sin}[\phi]^2 + \frac{1}{4}|0101\rangle\otimes|0101\rangle\,\mathrm{Sin}[\phi]^2 +$$
$$\frac{1}{4}|0110\rangle\otimes|0110\rangle\,\mathrm{Sin}[\phi]^2 + \frac{1}{4}|1001\rangle\otimes|1001\rangle\,\mathrm{Sin}[\phi]^2 + \frac{1}{4}|1010\rangle\otimes|1010\rangle\,\mathrm{Sin}[\phi]^2 + \frac{1}{4}|1100\rangle\otimes|1100\rangle\,\mathrm{Sin}[\phi]^2$$

If Alice (or Bob) measures the total Z, i.e., $Z = Z_1 + Z_2 + \ldots + Z_n$ on her qubits (see below), then the possible outcomes are $m = 0, 1, \ldots, n$ and the corresponding probabilities are given by the following formula.

In[168]:=

```
prb[ϕ_, m_] := With[{p = Sin[ϕ / 2] ^ 2}, Binomial[$n, m] * (1 - p) ^ ($n - m) p ^ m]
```

For fixed ϕ, the probability distribution looks like the following plot. The highest probability is for $m = n\sin^2(\phi/2)$ ($m = 3$ for $\phi = 2\pi/3$).

In[170]:=

```
BarChart[
 Table[prb[2 Pi/3, m], {m, $n}],
 ChartLabels -> Range[$n],
 Axes -> None, Frame -> True,
 FrameLabel -> {"m", "Probability"}
]
```

Out[170]=



Measurement of Total Pauli Z

How can we actually measure $Z_{\mathrm{tot}} := Z_1 + Z_2 + \ldots + Z_n$ on Bob's (or, equivalently, Alice's) qubits.

To specific, we assume $\phi = 2\pi/3$ from now on.

In[181]:=

```
ϕ = 2 Pi/3
```

Out[181]=

$$\frac{2\pi}{3}$$

This is a tiny utility function for convenience.

In[208]:=

```
bundle[] := Flatten@Table[bundle[c], {c, $n}]
bundle[c_] := With[
  {log = Ceiling@Log[2, c + 1]},
  Table[
   CNOT[Prepend[T@Range[k - 1], B[c]], T[k]],
   {k, log, 1, -1}
  ]
 ]
```

The value of $Z_{\mathtt{tot}}$ is stored on Teddy's register.

In[210]:=

```
qc1 = QuantumCircuit[
  Ket[TT],
  Sequence @@ bundle[],
  "Invisible" -> B[$n + 1 / 2]]
```

Out[210]=



Let us check that the above quantum circuit indeed gives the sum of the bit values of the Bob’s qubits.

In[211]:=

```
in = Basis[BB];
out = qc1 ** in;
```

In[213]:=

```
sum = FromDigits[#, 2] & /@ Reverse /@ Through[out[TT]];
Thread[SimpleForm[in] -> sum] // TableForm
```

Out[214]//TableForm=

```
|0000⟩ → 0
|0001⟩ → 1
|0010⟩ → 1
|0011⟩ → 2
|0100⟩ → 1
|0101⟩ → 2
|0110⟩ → 2
|0111⟩ → 3
|1000⟩ → 1
|1001⟩ → 2
|1010⟩ → 2
|1011⟩ → 3
|1100⟩ → 2
|1101⟩ → 3
|1110⟩ → 3
|1111⟩ → 4
```

In actual situation, one must perform the measurement on Teddy's register in the computational basis. The measurement outcome refers to the value of Z_{tot}.

In[215]:=

```
qc2 = QuantumCircuit[
  "Invisible" -> {A[$n + 1 / 2], B[$n + 1 / 2], T[5]},
  qc0, "Separator", qc1, "Spacer", Measurement[T[Range@ln, 3]]]
```

Out[215]=



Transformation

The post-measurement state is spread over the four qubits of Alice's and Bob's registers, respectively. One has to transform it into a smaller number of pairs. Here, we assume that the measurement outcome is $Z_{\text{tot}} = 3$, corresponding to two maximally entangled pairs. Note that measurement outcome $Z_{\text{tot}} = 1$ also leads to two maximally entangled pairs since Binomial [4, 1] == Binomial [4, 3] == 4.

To get the unitary transformation on Alice's side, we first rearrange the computational basis to get a new basis

$$\mathcal{A} := \left\{ \left|0111\right\rangle, \left|1011\right\rangle, \left|1101\right\rangle, \left|1110\right\rangle, \left|\alpha_5\right\rangle, \ldots, \left|\alpha_{16}\right\rangle \right\},$$

where $\left|\alpha_k\right\rangle$ ($k = 5, \ldots, 16$) are the computational basis states with total spin not equal to 3. Next, consider still another basis

$$\mathcal{A}' := \left\{ \left|00; 00\right\rangle, \left|01; 00\right\rangle, \left|10; 00\right\rangle, \left|11; 00\right\rangle, \left|\alpha'_5\right\rangle, \ldots, \left|\alpha'_{16}\right\rangle \right\},$$

where we have put a semicolon ';' to indicate the first two qubits to which encode Alice’s part of the maximally entangled states. Then, the required unitary transformation corresponds to the basis change from $\mathcal{A}$ to $\mathcal{A}'$. That is to say, the unitary transformation is given by

$$U := \left|00; 00\right\rangle\left\langle 0111\right| + \left|01; 00\right\rangle\left\langle 1011\right| + \left|10; 00\right\rangle\left\langle 1101\right| + \left|11; 00\right\rangle\left\langle 1110\right| + \sum_{k=5}^{16} \left|\alpha'_k\right\rangle\left\langle \alpha_k\right|.$$

The unitary transformation on Bob's register may be obtained in the same manner, where we denote the rearranged bases by $\mathcal{B}$ and $\mathcal{B}'$

Method 1: Using Dyad

These are the computational bases for Alice's and Bob's registers, respectively.

In[80]:=

```
bsA = Basis[AA]
bsB = Basis[BB]
```

Out[80]=

$$\{|\sqcup\rangle, |1_{A_4}\rangle, |1_{A_3}\rangle, |1_{A_3}1_{A_4}\rangle, |1_{A_2}\rangle, |1_{A_2}1_{A_4}\rangle, |1_{A_2}1_{A_3}\rangle, |1_{A_2}1_{A_3}1_{A_4}\rangle, |1_{A_1}\rangle, |1_{A_1}1_{A_4}\rangle, |1_{A_1}1_{A_3}\rangle, |1_{A_1}1_{A_3}1_{A_4}\rangle, |1_{A_1}1_{A_2}\rangle, |1_{A_1}1_{A_2}1_{A_4}\rangle, |1_{A_1}1_{A_2}1_{A_3}\rangle, |1_{A_1}1_{A_2}1_{A_3}1_{A_4}\rangle\}$$

Out[81]=

$$\{|\sqcup\rangle, |1_{B_4}\rangle, |1_{B_3}\rangle, |1_{B_3}1_{B_4}\rangle, |1_{B_2}\rangle, |1_{B_2}1_{B_4}\rangle, |1_{B_2}1_{B_3}\rangle, |1_{B_2}1_{B_3}1_{B_4}\rangle, |1_{B_1}\rangle, |1_{B_1}1_{B_4}\rangle, |1_{B_1}1_{B_3}\rangle, |1_{B_1}1_{B_3}1_{B_4}\rangle, |1_{B_1}1_{B_2}\rangle, |1_{B_1}1_{B_2}1_{B_4}\rangle, |1_{B_1}1_{B_2}1_{B_3}\rangle, |1_{B_1}1_{B_2}1_{B_3}1_{B_4}\rangle\}$$

These are the bases $\mathcal{A}$ and $\mathcal{B}$ for Alice's and Bob's registers, respectively.

In[82]:=

```
bb = Permutations[PadLeft[Table[1, 3], $n]];
sppA = Ket[AA -> #] & /@ bb;
sppB = Ket[BB -> #] & /@ bb;
sppA = Join[sppA, SortBy[Complement[bsA, sppA], LogicalForm[#, AA] &]];
sppB = Join[sppB, SortBy[Complement[bsB, sppB], LogicalForm[#, BB] &]];
```

In[87]:=

```
sppA // LogicalForm
sppB // LogicalForm
```

Out[87]=

$$\{|0_{A_1}1_{A_2}1_{A_3}1_{A_4}\rangle, |1_{A_1}0_{A_2}1_{A_3}1_{A_4}\rangle, |1_{A_1}1_{A_2}0_{A_3}1_{A_4}\rangle, |1_{A_1}1_{A_2}1_{A_3}0_{A_4}\rangle, |0_{A_1}0_{A_2}0_{A_3}0_{A_4}\rangle, |0_{A_1}0_{A_2}0_{A_3}1_{A_4}\rangle, |0_{A_1}0_{A_2}1_{A_3}0_{A_4}\rangle, |0_{A_1}0_{A_2}1_{A_3}1_{A_4}\rangle, |0_{A_1}1_{A_2}0_{A_3}0_{A_4}\rangle, |0_{A_1}1_{A_2}0_{A_3}1_{A_4}\rangle, |0_{A_1}1_{A_2}1_{A_3}0_{A_4}\rangle, |1_{A_1}0_{A_2}0_{A_3}0_{A_4}\rangle, |1_{A_1}0_{A_2}0_{A_3}1_{A_4}\rangle, |1_{A_1}0_{A_2}1_{A_3}0_{A_4}\rangle, |1_{A_1}1_{A_2}0_{A_3}0_{A_4}\rangle, |1_{A_1}1_{A_2}1_{A_3}1_{A_4}\rangle\}$$

Out[88]=

$$\{|0_{B_1}1_{B_2}1_{B_3}1_{B_4}\rangle, |1_{B_1}0_{B_2}1_{B_3}1_{B_4}\rangle, |1_{B_1}1_{B_2}0_{B_3}1_{B_4}\rangle, |1_{B_1}1_{B_2}1_{B_3}0_{B_4}\rangle, |0_{B_1}0_{B_2}0_{B_3}0_{B_4}\rangle, |0_{B_1}0_{B_2}0_{B_3}1_{B_4}\rangle, |0_{B_1}0_{B_2}1_{B_3}0_{B_4}\rangle, |0_{B_1}0_{B_2}1_{B_3}1_{B_4}\rangle, |0_{B_1}1_{B_2}0_{B_3}0_{B_4}\rangle, |0_{B_1}1_{B_2}0_{B_3}1_{B_4}\rangle, |0_{B_1}1_{B_2}1_{B_3}0_{B_4}\rangle, |1_{B_1}0_{B_2}0_{B_3}0_{B_4}\rangle, |1_{B_1}0_{B_2}0_{B_3}1_{B_4}\rangle, |1_{B_1}0_{B_2}1_{B_3}0_{B_4}\rangle, |1_{B_1}1_{B_2}0_{B_3}0_{B_4}\rangle, |1_{B_1}1_{B_2}1_{B_3}1_{B_4}\rangle\}$$

These are the bases $\mathcal{A}'$ and $\mathcal{B}'$ for Alice's and Bob's registers, respectively.

In[89]:=

```
newA = Basis[A@{1, 2}];
newB = Basis[B@{1, 2}];
newA = Join[newA, SortBy[Complement[bsA, newA], LogicalForm[#, AA] &]];
newB = Join[newB, SortBy[Complement[bsB, newB], LogicalForm[#, BB] &]];
```

In[93]:=

```
LogicalForm[newA]
LogicalForm[newB]
```

Out[93]=

$$\{|0_{A_1}0_{A_2}0_{A_3}0_{A_4}\rangle, |0_{A_1}1_{A_2}0_{A_3}0_{A_4}\rangle, |1_{A_1}0_{A_2}0_{A_3}0_{A_4}\rangle, |1_{A_1}1_{A_2}0_{A_3}0_{A_4}\rangle, |0_{A_1}0_{A_2}0_{A_3}1_{A_4}\rangle, |0_{A_1}0_{A_2}1_{A_3}0_{A_4}\rangle, |0_{A_1}0_{A_2}1_{A_3}1_{A_4}\rangle, |0_{A_1}1_{A_2}0_{A_3}1_{A_4}\rangle, |0_{A_1}1_{A_2}1_{A_3}0_{A_4}\rangle, |0_{A_1}1_{A_2}1_{A_3}1_{A_4}\rangle, |1_{A_1}0_{A_2}0_{A_3}1_{A_4}\rangle, |1_{A_1}0_{A_2}1_{A_3}0_{A_4}\rangle, |1_{A_1}0_{A_2}1_{A_3}1_{A_4}\rangle, |1_{A_1}1_{A_2}0_{A_3}1_{A_4}\rangle, |1_{A_1}1_{A_2}1_{A_3}0_{A_4}\rangle, |1_{A_1}1_{A_2}1_{A_3}1_{A_4}\rangle\}$$

Out[94]=

$$\{|0_{B_1}0_{B_2}0_{B_3}0_{B_4}\rangle, |0_{B_1}1_{B_2}0_{B_3}0_{B_4}\rangle, |1_{B_1}0_{B_2}0_{B_3}0_{B_4}\rangle, |1_{B_1}1_{B_2}0_{B_3}0_{B_4}\rangle, |0_{B_1}0_{B_2}0_{B_3}1_{B_4}\rangle, |0_{B_1}0_{B_2}1_{B_3}0_{B_4}\rangle, |0_{B_1}0_{B_2}1_{B_3}1_{B_4}\rangle, |0_{B_1}1_{B_2}0_{B_3}1_{B_4}\rangle, |0_{B_1}1_{B_2}1_{B_3}0_{B_4}\rangle, |0_{B_1}1_{B_2}1_{B_3}1_{B_4}\rangle, |1_{B_1}0_{B_2}0_{B_3}1_{B_4}\rangle, |1_{B_1}0_{B_2}1_{B_3}0_{B_4}\rangle, |1_{B_1}0_{B_2}1_{B_3}1_{B_4}\rangle, |1_{B_1}1_{B_2}0_{B_3}1_{B_4}\rangle, |1_{B_1}1_{B_2}1_{B_3}0_{B_4}\rangle, |1_{B_1}1_{B_2}1_{B_3}1_{B_4}\rangle\}$$

Finally, the unitary transformation is constructed as prescribed above.

In[95]:=

```
trsA = Total@MapThread[Dyad[#1, #2, AA] &, {newA, sppA}]
trsB = Total@MapThread[Dyad[#1, #2, BB] &, {newB, sppB}]
```

Out[95]=

$$\left|0_{A_1}0_{A_2}0_{A_3}0_{A_4}\right\rangle\left\langle 0_{A_1}1_{A_2}1_{A_3}1_{A_4}\right| + \left|1_{A_1}0_{A_2}0_{A_3}0_{A_4}\right\rangle\left\langle 1_{A_1}1_{A_2}0_{A_3}1_{A_4}\right| + \left|1_{A_1}1_{A_2}0_{A_3}0_{A_4}\right\rangle\left\langle 1_{A_1}1_{A_2}1_{A_3}0_{A_4}\right| + \left|1_{A_1}1_{A_2}1_{A_3}0_{A_4}\right\rangle\left\langle 1_{A_1}1_{A_2}0_{A_3}0_{A_4}\right| +$$
$$\left|1_{A_1}1_{A_2}1_{A_3}1_{A_4}\right\rangle\left\langle 1_{A_1}1_{A_2}1_{A_3}1_{A_4}\right| + \left|1_{A_1}1_{A_2}0_{A_3}1_{A_4}\right\rangle\left\langle 1_{A_1}0_{A_2}1_{A_3}0_{A_4}\right| + \left|1_{A_1}0_{A_2}1_{A_3}0_{A_4}\right\rangle\left\langle 1_{A_1}0_{A_2}0_{A_3}0_{A_4}\right| + \left|1_{A_1}0_{A_2}1_{A_3}1_{A_4}\right\rangle\left\langle 1_{A_1}0_{A_2}0_{A_3}1_{A_4}\right| +$$
$$\left|1_{A_1}0_{A_2}0_{A_3}1_{A_4}\right\rangle\left\langle 0_{A_1}1_{A_2}1_{A_3}0_{A_4}\right| + \left|0_{A_1}1_{A_2}0_{A_3}0_{A_4}\right\rangle\left\langle 1_{A_1}0_{A_2}1_{A_3}1_{A_4}\right| + \left|0_{A_1}1_{A_2}1_{A_3}0_{A_4}\right\rangle\left\langle 0_{A_1}1_{A_2}0_{A_3}0_{A_4}\right| + \left|0_{A_1}1_{A_2}1_{A_3}1_{A_4}\right\rangle\left\langle 0_{A_1}1_{A_2}0_{A_3}1_{A_4}\right| +$$
$$\left|0_{A_1}1_{A_2}0_{A_3}1_{A_4}\right\rangle\left\langle 0_{A_1}0_{A_2}1_{A_3}1_{A_4}\right| + \left|0_{A_1}0_{A_2}1_{A_3}0_{A_4}\right\rangle\left\langle 0_{A_1}0_{A_2}0_{A_3}1_{A_4}\right| + \left|0_{A_1}0_{A_2}1_{A_3}1_{A_4}\right\rangle\left\langle 0_{A_1}0_{A_2}1_{A_3}0_{A_4}\right| + \left|0_{A_1}0_{A_2}0_{A_3}1_{A_4}\right\rangle\left\langle 0_{A_1}0_{A_2}0_{A_3}0_{A_4}\right|$$

Out[96]=

$$\left|0_{B_1}0_{B_2}0_{B_3}0_{B_4}\right\rangle\left\langle 0_{B_1}1_{B_2}1_{B_3}1_{B_4}\right| + \left|1_{B_1}0_{B_2}0_{B_3}0_{B_4}\right\rangle\left\langle 1_{B_1}1_{B_2}0_{B_3}1_{B_4}\right| + \left|1_{B_1}1_{B_2}0_{B_3}0_{B_4}\right\rangle\left\langle 1_{B_1}1_{B_2}1_{B_3}0_{B_4}\right| + \left|1_{B_1}1_{B_2}1_{B_3}0_{B_4}\right\rangle\left\langle 1_{B_1}1_{B_2}0_{B_3}0_{B_4}\right| +$$
$$\left|1_{B_1}1_{B_2}1_{B_3}1_{B_4}\right\rangle\left\langle 1_{B_1}1_{B_2}1_{B_3}1_{B_4}\right| + \left|1_{B_1}1_{B_2}0_{B_3}1_{B_4}\right\rangle\left\langle 1_{B_1}0_{B_2}1_{B_3}0_{B_4}\right| + \left|1_{B_1}0_{B_2}1_{B_3}0_{B_4}\right\rangle\left\langle 1_{B_1}0_{B_2}0_{B_3}0_{B_4}\right| + \left|1_{B_1}0_{B_2}1_{B_3}1_{B_4}\right\rangle\left\langle 1_{B_1}0_{B_2}0_{B_3}1_{B_4}\right| +$$
$$\left|1_{B_1}0_{B_2}0_{B_3}1_{B_4}\right\rangle\left\langle 0_{B_1}1_{B_2}1_{B_3}0_{B_4}\right| + \left|0_{B_1}1_{B_2}0_{B_3}0_{B_4}\right\rangle\left\langle 1_{B_1}0_{B_2}1_{B_3}1_{B_4}\right| + \left|0_{B_1}1_{B_2}1_{B_3}0_{B_4}\right\rangle\left\langle 0_{B_1}1_{B_2}0_{B_3}0_{B_4}\right| + \left|0_{B_1}1_{B_2}1_{B_3}1_{B_4}\right\rangle\left\langle 0_{B_1}1_{B_2}0_{B_3}1_{B_4}\right| +$$
$$\left|0_{B_1}1_{B_2}0_{B_3}1_{B_4}\right\rangle\left\langle 0_{B_1}0_{B_2}1_{B_3}1_{B_4}\right| + \left|0_{B_1}0_{B_2}1_{B_3}0_{B_4}\right\rangle\left\langle 0_{B_1}0_{B_2}0_{B_3}1_{B_4}\right| + \left|0_{B_1}0_{B_2}1_{B_3}1_{B_4}\right\rangle\left\langle 0_{B_1}0_{B_2}1_{B_3}0_{B_4}\right| + \left|0_{B_1}0_{B_2}0_{B_3}1_{B_4}\right\rangle\left\langle 0_{B_1}0_{B_2}0_{B_3}0_{B_4}\right|$$

By construction, the unitary transformations are just permutation, which is clear from their matrix representations.

In[97]:=

```
Matrix[trsA] // MatrixForm
Matrix[trsB] // MatrixForm
```

Out[97]//MatrixForm=

$$\begin{pmatrix}
0&0&0&0&0&0&0&1&0&0&0&0&0&0&0&0\\
1&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&1&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&1&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&1&0&0&0&0\\
0&0&0&1&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&1&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&1&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&1&0&0\\
0&0&0&0&0&0&1&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&1&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&1&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&1&0\\
0&0&0&0&0&0&0&0&0&0&1&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&1&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&1
\end{pmatrix}$$

Out[98]//MatrixForm=

$$\begin{pmatrix}
0&0&0&0&0&0&0&1&0&0&0&0&0&0&0&0\\
1&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&1&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&1&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&1&0&0&0&0\\
0&0&0&1&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&1&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&1&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&1&0&0\\
0&0&0&0&0&0&1&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&1&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&1&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&1&0\\
0&0&0&0&0&0&0&0&0&0&1&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&1&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&1
\end{pmatrix}$$

Check that they are indeed unitary transformations.

In[47]:=

```
Dagger[trsA] ** trsA // Elaborate
Dagger[trsB] ** trsB // Elaborate
```

Out[47]=

```
1
```

Out[48]=

```
1
```

Method 2: Using Permutation

Here, we find the permutation corresponding to the required basis change and use it to construct the unitary matrix.

In[34]:=

```
conststructUnitary::usage =
  "contructUnitary[m, n] finds the permutation of computational basis states of n
    qubits, where states with total spin m are mapped to the standard computational
    basis states of the first k qubits with other qubits set to zero. Here,
    k=Log[2,Binomial[n,m]]. It returns the matrix representing the permutation.";
```

In[35]:=

```
constructUnitary::bad = "States with `` non-zero qubits in
    a register of `` qubits cannot be encoded in a finite number of qubits.";
```

In[36]:=

```
constructUnitary[m_Integer, n_Integer] := Module[
  {org = Tuples[{0, 1}, n],
   bin = Log[2, Binomial[n, m]],
   src, dst},
  If[Not@IntegerQ@bin,
   Message[constructUnitary::bad, m, n];
   Return[One@Power[2, n]]
  ];
  src = Permutations@PadLeft[Table[1, m], n];
  src = DeleteDuplicates@Join[src, org];
  dst = PadRight[#, n] & /@ Tuples[{0, 1}, bin];
  dst = DeleteDuplicates@Join[dst, org];
  tau = FindPermutation[src, org];
  tau = FindPermutation@Permute[dst, tau];
  PermutationMatrix[tau, Power[2, n]]
 ]
```

Take an example for $m = 3$. Each row or column contains exactly one element of value 1 and all other elements are zero, indicating the matrix represents a permutation.

In[15]:=

```
mat = constructUnitary[3, 4];
mat // MatrixForm
```

Out[16]//MatrixForm=

$$\begin{pmatrix}
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1
\end{pmatrix}$$

Construct the operators corresponding to the above matrix.

In[17]:=

```
trsA = ExpressionFor[mat, AA] // Elaborate;
trsB = ExpressionFor[mat, BB] // Elaborate;
```

Examine if the operator properly maps the computational basis states for Alice's qubits.

In[25]:=

```
in = Basis[AA];
out = trsA ** in;
Thread[in -> out] // LogicalForm // TableForm
```

Out[27]//TableForm=

$\left|0_{A_1}0_{A_2}0_{A_3}0_{A_4}\right\rangle \to \left|0_{A_1}0_{A_2}0_{A_3}1_{A_4}\right\rangle$

$\left|0_{A_1}0_{A_2}0_{A_3}1_{A_4}\right\rangle \to \left|0_{A_1}0_{A_2}1_{A_3}0_{A_4}\right\rangle$

$\left|0_{A_1}0_{A_2}1_{A_3}0_{A_4}\right\rangle \to \left|0_{A_1}0_{A_2}1_{A_3}1_{A_4}\right\rangle$

$\left|0_{A_1}0_{A_2}1_{A_3}1_{A_4}\right\rangle \to \left|0_{A_1}1_{A_2}0_{A_3}1_{A_4}\right\rangle$

$\left|0_{A_1}1_{A_2}0_{A_3}0_{A_4}\right\rangle \to \left|0_{A_1}1_{A_2}1_{A_3}0_{A_4}\right\rangle$

$\left|0_{A_1}1_{A_2}0_{A_3}1_{A_4}\right\rangle \to \left|0_{A_1}1_{A_2}1_{A_3}1_{A_4}\right\rangle$

$\left|0_{A_1}1_{A_2}1_{A_3}0_{A_4}\right\rangle \to \left|1_{A_1}0_{A_2}0_{A_3}1_{A_4}\right\rangle$

$\left|0_{A_1}1_{A_2}1_{A_3}1_{A_4}\right\rangle \to \left|0_{A_1}0_{A_2}0_{A_3}0_{A_4}\right\rangle$

$\left|1_{A_1}0_{A_2}0_{A_3}0_{A_4}\right\rangle \to \left|1_{A_1}0_{A_2}1_{A_3}0_{A_4}\right\rangle$

$\left|1_{A_1}0_{A_2}0_{A_3}1_{A_4}\right\rangle \to \left|1_{A_1}0_{A_2}1_{A_3}1_{A_4}\right\rangle$

$\left|1_{A_1}0_{A_2}1_{A_3}0_{A_4}\right\rangle \to \left|1_{A_1}1_{A_2}0_{A_3}1_{A_4}\right\rangle$

$\left|1_{A_1}0_{A_2}1_{A_3}1_{A_4}\right\rangle \to \left|0_{A_1}1_{A_2}0_{A_3}0_{A_4}\right\rangle$

$\left|1_{A_1}1_{A_2}0_{A_3}0_{A_4}\right\rangle \to \left|1_{A_1}1_{A_2}1_{A_3}0_{A_4}\right\rangle$

$\left|1_{A_1}1_{A_2}0_{A_3}1_{A_4}\right\rangle \to \left|1_{A_1}0_{A_2}0_{A_3}0_{A_4}\right\rangle$

$\left|1_{A_1}1_{A_2}1_{A_3}0_{A_4}\right\rangle \to \left|1_{A_1}1_{A_2}0_{A_3}0_{A_4}\right\rangle$

$\left|1_{A_1}1_{A_2}1_{A_3}1_{A_4}\right\rangle \to \left|1_{A_1}1_{A_2}1_{A_3}1_{A_4}\right\rangle$

Also examine if the operator properly maps the computational basis states for Bob's qubits.

In[28]:=

```
in = Basis[BB];
out = trsB ** in;
Thread[in -> out] // LogicalForm // TableForm
```

Out[30]//TableForm=

$\left|0_{B_1}0_{B_2}0_{B_3}0_{B_4}\right\rangle \to \left|0_{B_1}0_{B_2}0_{B_3}1_{B_4}\right\rangle$

$\left|0_{B_1}0_{B_2}0_{B_3}1_{B_4}\right\rangle \to \left|0_{B_1}0_{B_2}1_{B_3}0_{B_4}\right\rangle$

$\left|0_{B_1}0_{B_2}1_{B_3}0_{B_4}\right\rangle \to \left|0_{B_1}0_{B_2}1_{B_3}1_{B_4}\right\rangle$

$\left|0_{B_1}0_{B_2}1_{B_3}1_{B_4}\right\rangle \to \left|0_{B_1}1_{B_2}0_{B_3}1_{B_4}\right\rangle$

$\left|0_{B_1}1_{B_2}0_{B_3}0_{B_4}\right\rangle \to \left|0_{B_1}1_{B_2}1_{B_3}0_{B_4}\right\rangle$

$\left|0_{B_1}1_{B_2}0_{B_3}1_{B_4}\right\rangle \to \left|0_{B_1}1_{B_2}1_{B_3}1_{B_4}\right\rangle$

$\left|0_{B_1}1_{B_2}1_{B_3}0_{B_4}\right\rangle \to \left|1_{B_1}0_{B_2}0_{B_3}1_{B_4}\right\rangle$

$\left|0_{B_1}1_{B_2}1_{B_3}1_{B_4}\right\rangle \to \left|0_{B_1}0_{B_2}0_{B_3}0_{B_4}\right\rangle$

$\left|1_{B_1}0_{B_2}0_{B_3}0_{B_4}\right\rangle \to \left|1_{B_1}0_{B_2}1_{B_3}0_{B_4}\right\rangle$

$\left|1_{B_1}0_{B_2}0_{B_3}1_{B_4}\right\rangle \to \left|1_{B_1}0_{B_2}1_{B_3}1_{B_4}\right\rangle$

$\left|1_{B_1}0_{B_2}1_{B_3}0_{B_4}\right\rangle \to \left|1_{B_1}1_{B_2}0_{B_3}1_{B_4}\right\rangle$

$\left|1_{B_1}0_{B_2}1_{B_3}1_{B_4}\right\rangle \to \left|0_{B_1}1_{B_2}0_{B_3}0_{B_4}\right\rangle$

$\left|1_{B_1}1_{B_2}0_{B_3}0_{B_4}\right\rangle \to \left|1_{B_1}1_{B_2}1_{B_3}0_{B_4}\right\rangle$

$\left|1_{B_1}1_{B_2}0_{B_3}1_{B_4}\right\rangle \to \left|1_{B_1}0_{B_2}0_{B_3}0_{B_4}\right\rangle$

$\left|1_{B_1}1_{B_2}1_{B_3}0_{B_4}\right\rangle \to \left|1_{B_1}1_{B_2}0_{B_3}0_{B_4}\right\rangle$

$\left|1_{B_1}1_{B_2}1_{B_3}1_{B_4}\right\rangle \to \left|1_{B_1}1_{B_2}1_{B_3}1_{B_4}\right\rangle$

Overall

Now, we have all components. Our desired quantum circuit is as follows. The first part generates four partially entangled pairs, the second measures the total Pauli Z on Bob's register, and the last transforms the post-measurement state a fixed pair of qubits from Alice's and Bob's registers.

In[35]:=

```
all = QuantumCircuit[qc2, "Separator",
  ControlledU[T@{1, 2, 3} -> {1, 1, 0}, trsA, "Label" -> "U"],
  ControlledU[T@{1, 2, 3} -> {1, 1, 0}, trsB, "Label" -> "U"],
 ]
```

Out[35]=



Because the transformation on Alice’s and Bob’s sides are designed for the measurement outcome of $Z_{\text{tot}} = 3$, we just run the quantum circuit until we get the desired outcome.

In[36]:=

```
Until[ztot == 3,
  out = Elaborate[all];
  ztot = FromDigits[Reverse@Readout[Through[TT[3]]], 2];
  PrintTemporary["Ztot=", ztot]
 ];
```

Here is the output state of the total system. Obviously, the whole register *T* is separated from the other two registers. So are the last two registers of the respective registers A and B.

In[37]:=

```
LogicalForm[out, Join[AA, BB, TT]]
```

Out[37]=

$$\frac{1}{2}\left|0_{A_1}0_{A_2}0_{A_3}0_{A_4}0_{B_1}0_{B_2}0_{B_3}0_{B_4}1_{T_1}1_{T_2}0_{T_3}\right\rangle + \frac{1}{2}\left|0_{A_1}1_{A_2}0_{A_3}0_{A_4}0_{B_1}1_{B_2}0_{B_3}0_{B_4}1_{T_1}1_{T_2}0_{T_3}\right\rangle + \frac{1}{2}\left|1_{A_1}0_{A_2}0_{A_3}0_{A_4}1_{B_1}0_{B_2}0_{B_3}0_{B_4}1_{T_1}1_{T_2}0_{T_3}\right\rangle + \frac{1}{2}\left|1_{A_1}1_{A_2}0_{A_3}0_{A_4}1_{B_1}1_{B_2}0_{B_3}0_{B_4}1_{T_1}1_{T_2}0_{T_3}\right\rangle$$

Focusing on the first two qubits of registers A and B, we see that there are two maximally entangled pairs.

In[38]:=

```
SimpleForm[out, {A@{1, 2}, B@{1, 2}}]
```

Out[38]=

$$\frac{|00;00\rangle}{2}+\frac{|01;01\rangle}{2}+\frac{|10;10\rangle}{2}+\frac{|11;11\rangle}{2}$$

In[39]:=

```
KetFactor@KetDrop[out, TT]
```

Out[39]=

$$\frac{1}{2}\left(\left|0_{A_1}0_{B_1}\right\rangle+\left|1_{A_1}1_{B_1}\right\rangle\right)\otimes\left(\left|0_{A_2}0_{B_2}\right\rangle+\left|1_{A_2}1_{B_2}\right\rangle\right)$$

Remark

So, we have successfully generated two maximally entangled pairs from four partially entangled pairs. Note that this was possible because we assumed that the measurement outcome was $m = 3$. The case with $m = 1$ is also handled following a similar procedure.

However, if the measurement outcome is $m = 2$, resulting in the post-measurement state

$$\frac{1}{6}|0011\rangle\otimes|0011\rangle+\frac{1}{6}|0101\rangle\otimes|0101\rangle+\frac{1}{6}|0110\rangle\otimes|0110\rangle+\frac{1}{6}|1001\rangle\otimes|1001\rangle+\frac{1}{6}|1010\rangle\otimes|1010\rangle+\frac{1}{6}|1100\rangle\otimes|1100\rangle$$

then there is no way to turn this state into a product state on a system of qubits.

Related Guides

- Quantum Information Systems
- Quantum Many-Body Systems
- Quantum Spin Systems

Related Tech Notes

- A Quantum Playbook
- Entanglement Distillation
- Quantum Information Theory
- Quantum Information Systems with Q3
- Quantum Many-Body Systems with Q3
- Quantum Spin Systems with Q3

Related Links

- J. Preskill (1998) , Lecture Notes for Physics 229: Quantum Information and Computation.
- M. Nielsen and I. L. Chuang (2022) , Quantum Computation and Quantum Information (Cambridge University Press, 2011).
- Mahn-Soo Choi (2022) , A Quantum Computation Workbook (Springer, 2022).