The Controlled-Z (CZ) Gate

In[ ]:=
```
Let[Qubit, S]
```

What is it?

CNOT: Revisited

In[ ]:=
```
cnot = QuantumCircuit[CNOT[S[1], S[2]]]
```

Out[ ]=



In[ ]:=
```
in = Basis[S@{1, 2}];
out = cnot ** in;
ProductForm[Thread[in → out], S@{1, 2}] // TableForm
```

Out[ ]//TableForm=

$$|0\rangle \otimes |0\rangle \to |0\rangle \otimes |0\rangle$$
$$|0\rangle \otimes |1\rangle \to |0\rangle \otimes |1\rangle$$
$$|1\rangle \otimes |0\rangle \to |1\rangle \otimes |1\rangle$$
$$|1\rangle \otimes |1\rangle \to |1\rangle \otimes |0\rangle$$

In[ ]:=
```
Matrix[cnot] // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

CZ

In[ ]:=
```
cz = QuantumCircuit[CZ[S[1], S[2]]]
```

Out[ ]=



In[ ]:=
```
in = Basis[S@{1, 2}];
out = cz ** in;
ProductForm[Thread[in → out], S@{1, 2}] // TableForm
```

Out[ ]//TableForm=

$$
\begin{array}{l}
|0\rangle \otimes |0\rangle \to |0\rangle \otimes |0\rangle \\
|0\rangle \otimes |1\rangle \to |0\rangle \otimes |1\rangle \\
|1\rangle \otimes |0\rangle \to |1\rangle \otimes |0\rangle \\
|1\rangle \otimes |1\rangle \to -|1\rangle \otimes |1\rangle
\end{array}
$$

In[ ]:=
```
Matrix[cz] // MatrixForm
```

Out[ ]//MatrixForm=

$$
\begin{pmatrix}
1 & 0 & 0 & 0 \\
0 & 1 & 0 & 0 \\
0 & 0 & 1 & 0 \\
0 & 0 & 0 & -1
\end{pmatrix}
$$

Advantages of CZ

- Easier to physically realize in many cases.

Application: Graph State

In[ ]:=
```
links = Chain[FlavorNone@S@{1, 2, 3, 4}]
```

Out[ ]=

$\{S_1 \to S_2, S_2 \to S_3, S_3 \to S_4\}$

In[ ]:=
```
g = Graph[links]
```

Out[ ]=

In[ ]:=
```
qc = QuantumCircuit[Ket[S@{1, 2, 3, 4}], g]
```

Out[ ]=



In[•]:=
```
GraphState[g]
```

Out[•]=

$$\frac{1}{4}\left|0_{S_1}0_{S_2}0_{S_3}0_{S_4}\right\rangle+\frac{1}{4}\left|0_{S_1}0_{S_2}0_{S_3}1_{S_4}\right\rangle+\frac{1}{4}\left|0_{S_1}0_{S_2}1_{S_3}0_{S_4}\right\rangle-\frac{1}{4}\left|0_{S_1}0_{S_2}1_{S_3}1_{S_4}\right\rangle+$$
$$\frac{1}{4}\left|0_{S_1}1_{S_2}0_{S_3}0_{S_4}\right\rangle+\frac{1}{4}\left|0_{S_1}1_{S_2}0_{S_3}1_{S_4}\right\rangle-\frac{1}{4}\left|0_{S_1}1_{S_2}1_{S_3}0_{S_4}\right\rangle+\frac{1}{4}\left|0_{S_1}1_{S_2}1_{S_3}1_{S_4}\right\rangle+$$
$$\frac{1}{4}\left|1_{S_1}0_{S_2}0_{S_3}0_{S_4}\right\rangle+\frac{1}{4}\left|1_{S_1}0_{S_2}0_{S_3}1_{S_4}\right\rangle+\frac{1}{4}\left|1_{S_1}0_{S_2}1_{S_3}0_{S_4}\right\rangle-\frac{1}{4}\left|1_{S_1}0_{S_2}1_{S_3}1_{S_4}\right\rangle-$$
$$\frac{1}{4}\left|1_{S_1}1_{S_2}0_{S_3}0_{S_4}\right\rangle-\frac{1}{4}\left|1_{S_1}1_{S_2}0_{S_3}1_{S_4}\right\rangle+\frac{1}{4}\left|1_{S_1}1_{S_2}1_{S_3}0_{S_4}\right\rangle-\frac{1}{4}\left|1_{S_1}1_{S_2}1_{S_3}1_{S_4}\right\rangle$$

CZ vs CNOT

Recall the identity.

In[•]:=
```
S[6] ** S[1] ** S[6] // Elaborate
```

Out[•]=

S^z

Keeping the identity in mind, let us stat with the CNOT gate.

In[•]:=
```
cnot = QuantumCircuit[CNOT[S[1], S[2]]]
```

Out[•]=



Using the identity above, we can see that the following to quantum circuits are equivalent.

In[•]:=
```
Row@{
  new = QuantumCircuit[S[2, 6], cnot, S[2, 6]], "=",
  cz = QuantumCircuit[CZ[S[1], S[2]]]
 }
```

Out[•]=



Indeed, check the matrix representation of the second quantum circuit.

In[•]:=
```
Matrix[new] // MatrixForm
```

Out[•]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

In[ ]:=
```
more = QuantumCircuit[S[2, 6], CZ[S[1], S[2]], S[2, 6]]
```
Out[ ]=



In[ ]:=
```
more - cnot // Elaborate
```
Out[ ]=
```
0
```

In[ ]:=
```
QuantumCircuit[S[2, 6], S[2, 6], cnot, S[2, 6], S[2, 6]]
```
Out[ ]=



In[ ]:=
```
back = QuantumCircuit[S[2, 6], S[1, 6], CNOT[S[2], S[1]], S[1, 6], S[2, 6]]
```
Out[ ]=



In[ ]:=
```
Row@{cnot, "=",
   QuantumCircuit[S[{1, 2}, 6], CNOT[S[2], S[1]], S[{1, 2}, 6]]
  }
```
Out[ ]=



SWAP Gate

In[ ]:=
```
swap = QuantumCircuit[SWAP[S[1], S[2]]]
```
Out[ ]=



In[ ]:=
```
in = Basis[S@{1, 2}];
out = swap ** in;
ProductForm[Thread[in → out], S@{1, 2}] // TableForm
```
Out[ ]//TableForm=

$|0\rangle \otimes |0\rangle \rightarrow |0\rangle \otimes |0\rangle$
$|0\rangle \otimes |1\rangle \rightarrow |1\rangle \otimes |0\rangle$
$|1\rangle \otimes |0\rangle \rightarrow |0\rangle \otimes |1\rangle$
$|1\rangle \otimes |1\rangle \rightarrow |1\rangle \otimes |1\rangle$

In[ ]:=
```
Matrix[swap] // MatrixForm
```
Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Multi-Control Z Gate

In[ ]:=
```
QuantumCircuit[CZ[S[{1, 2, 3}, $], S[4]]]
```
Out[ ]=



In[ ]:=
```
$n = 2;
cc = S[Range[$n], $];
tt = S[$n + 1, $];
cnot = QuantumCircuit[CZ[cc, tt]]
```
Out[ ]=



In[∘]:=
```
Matrix[cnot] // MatrixForm
```
Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 \end{pmatrix}$$

In[∘]:=
```
in = Basis[cc, tt];
out = cnot ** in;
```

In[∘]:=
```
ProductForm[Thread[in → out], {cc, tt}] // TableForm
```
Out[∘]//TableForm=

$$\begin{array}{l} |00\rangle \otimes |0\rangle \to |00\rangle \otimes |0\rangle \\ |00\rangle \otimes |1\rangle \to |00\rangle \otimes |1\rangle \\ |01\rangle \otimes |0\rangle \to |01\rangle \otimes |0\rangle \\ |01\rangle \otimes |1\rangle \to |01\rangle \otimes |1\rangle \\ |10\rangle \otimes |0\rangle \to |10\rangle \otimes |0\rangle \\ |10\rangle \otimes |1\rangle \to |10\rangle \otimes |1\rangle \\ |11\rangle \otimes |0\rangle \to |11\rangle \otimes |0\rangle \\ |11\rangle \otimes |1\rangle \to -|11\rangle \otimes |1\rangle \end{array}$$

Summary

Functions

- `CZ vs CNOT`
- `Chain`
- `SWAP`
- `GraphState`

Related Links

- Chapters 2 of the Quantum Workbook (2022, 2023).
- Tutorial: “Quantum Computation: Overview”